

# Implementing Density Functional Theory

Fadjar Fathurrahman  
Hermawan Kresno Dipojono

June 26, 2020



# Preface

Importance of density functional theory

Implementation of density functional theory in various program packages (free and commercial)

Several books about density functional theories

The problem: not yet giving necessary details

This book is our humble attempt to demystifying several aspects of practical density functional theory to beginners in the field.

Objective of this book: show the reader how to implement a density functional theory for simple system containing only model potential (such as harmonic potential) and to non-local pseudopotentials which are usually used in typical DFT calculations for molecular and crystalline systems.

Outline of the book: 1d, 2d, 3d, Schrodinger equation, Poisson equation, Kohn-Sham equation for local (pseudo)potentials, Kohn-Sham equation for nonlocal pseudopotentials.

This is for *acknowledgments*.

Bandung,  
month year

*Fadjar Fathurrahman*  
*Hermawan Kresno Dipojono*



# Contents

|          |   |           |
|----------|---|-----------|
| <b>1</b> | <b>Introduction</b>   | <b>1</b>  |
| 1.1      | Density functional theory . . . . .                                       | 1         |
| <b>2</b> | <b>Schroedinger equation in 1d</b>  | <b>5</b>  |
| 2.1      | Grid points . . . . .   | 5         |
| 2.2      | Approximating second derivative operator . . . . .                        | 8         |
| 2.3      | Harmonic potential . . . . .  | 11        |
| 2.4      | Higher order finite difference . . . . .                                  | 14        |
| 2.5      | Harmonic potential with higher order finite-difference formulas . . . . . | 14        |
| 2.6      | Exercises . . . . .   | 16        |
| <b>3</b> | <b>Schroedinger equation in 2d</b>  | <b>17</b> |
| 3.1      | Describing grid in 2d . . . . .   | 17        |
| 3.2      | Laplacian operator . . . . .  | 19        |
| 3.3      | More about sparse matrices in Julia . . . . .                             | 21        |
| 3.4      | Iterative methods for eigenvalue problem . . . . .                        | 23        |
| 3.5      | Minimization approach: 2d harmonic potential . . . . .                    | 24        |
| 3.5.1    | Orthonormalization . . . . .  | 24        |
| 3.5.2    | Band energy functional and its gradient . . . . .                         | 25        |
| 3.5.3    | Steepest descent method . . . . .   | 25        |
| 3.5.4    | Line minimization . . . . .   | 27        |
| 3.5.5    | Preconditioning . . . . .   | 27        |
| 3.5.6    | Conjugate gradient . . . . .  | 29        |
| 3.5.7    | Eigenfunctions . . . . .  | 29        |
| 3.6      | Exercises . . . . .   | 30        |
| <b>4</b> | <b>Schroedinger equation in 3d</b>  | <b>31</b> |
| 4.0.1    | Three-dimensional grid . . . . .  | 31        |
| 4.0.2    | Laplacian operator . . . . .  | 31        |
| 4.1      | 3d harmonic oscillator . . . . .  | 32        |
| 4.2      | Hydrogen atom . . . . .   | 32        |
| 4.2.1    | Coulomb potential and its singularity . . . . .                           | 32        |

|          |   |           |
|----------|---|-----------|
| 4.2.2    | Pseudopotential . . . . .                         | 33        |
| 4.3      | Exercises . . . . .                               | 35        |
| <b>5</b> | <b>Poisson equation</b>                           | <b>37</b> |
| 5.1      | Conjugate gradient method . . . . .               | 37        |
| 5.2      | Reciprocal space method . . . . .                 | 39        |
| 5.3      | Direct integration . . . . .                      | 39        |
| <b>6</b> | <b>Kohn-Sham equation part I</b>                  | <b>41</b> |
| 6.1      | Hartree calculation . . . . .                     | 42        |
| 6.1.1    | The Hamiltonian struct . . . . .                  | 42        |
| 6.1.2    | Electrons . . . . .                               | 44        |
| 6.1.3    | Total energy terms . . . . .                      | 44        |
| 6.1.4    | Harmonic potential . . . . .                      | 46        |
| 6.1.5    | H atom . . . . .                                  | 48        |
| 6.1.6    | H <sub>2</sub> molecule . . . . .                 | 48        |
| 6.2      | Kohn-Sham calculations . . . . .                  | 49        |
| 6.2.1    | Exchange term: Slater exchange . . . . .          | 49        |
| 6.2.2    | Correlation VWN . . . . .                         | 49        |
| 6.2.3    | Implementation . . . . .                          | 50        |
| 6.2.4    | Harmonic potential . . . . .                      | 51        |
| 6.2.5    | Hydrogen atom . . . . .                           | 52        |
| 6.2.6    | H <sub>2</sub> molecule . . . . .                 | 52        |
| <b>7</b> | <b>Kohn-Sham equation part II</b>                 | <b>55</b> |
| <b>A</b> | <b>Introduction to Julia programming language</b> | <b>57</b> |
| A.1      | Installation . . . . .                            | 57        |
| A.2      | Using Julia . . . . .                             | 57        |
| A.2.1    | Using Julia REPL . . . . .                        | 57        |
| A.2.2    | Julia script file . . . . .                       | 58        |
| A.3      | Basic programming constructs . . . . .            | 59        |
| A.3.1    | Displaying something . . . . .                    | 59        |
| A.3.2    | Variables . . . . .                               | 59        |
| A.3.3    | Mathematical operators . . . . .                  | 60        |
| <b>B</b> | <b>Lagrange basis functions</b>                   | <b>61</b> |
| B.1      | Lagrange-sinc function . . . . .                  | 61        |
| B.2      | Periodic Lagrange function . . . . .              | 61        |
| B.3      | Cluster Lagrange function . . . . .               | 63        |
| <b>C</b> | <b>Introduction to Octopus DFT code</b>           | <b>65</b> |

# Chapter 1

## Introduction

### 1.1 Density functional theory

This section is intended as a brief overview of Kohn-Sham density functional theory. For more information, the reader is referred to the original articles [1, 2] and several books [3, 4, 5, 6].

Matters can be modeled as a collection of interacting electrons and nuclei. As such electrons shall be treated quantum mechanically while the nuclei can be considered as distributed points of positive charges. Surely, this collection forms a system with very complex internal interactions. Nevertheless the total energy of the system can simply be decomposed into kinetic and potential energies. The distributed nuclei of positive charges form Coulombic potential field affecting the interaction among the electrons in the system. Obviously, the electrons interaction will also affects the configuration of distributed nuclei. Deriving and solving an exact mathematical model of the system is an unforeseeable future right now. Fortunately, by relaxing the exactness through some approximations this very complex system can be treated with ease and still provides very useful results. Since the mass of nuclei is about three order of magnitude heavier than the mass of electrons, the Born-Oppenheimer suggests an approximation that the nuclei are relatively fixed to fastly moving electrons. This approximation transforms the complex problems of interacting nuclei and electrons into problems of solely interacting many electrons in an external fields. For the interacting electrons, the nuclei Coulombic potential field performs as external potential. Nonetheless looking for the exact solution wave functions for this many electrons problem is an impossible task. Wonderfully, with further approximations, computational or numerical methods provides the tractable ways to solve this extremely difficult problem. Kohn-Sham density functional theory [1, 2] is an attempt to offer useful model and computationally tractable solutions to this very complex system. The breakthrough provided by the Kohn-Sham approach is expressing the total energy as a functional of electrons density only. Marvelously the problem of looking for wave functions of many electrons problem becomes the much easier looking for electrons density.

In Kohn-Sham density functional theory, we are interested in total energy of a system of interacting electrons under external potential  $V_{\text{ext}}(\mathbf{r})$ . This quantity can be written as functional of a set of single-particle wave functions or Kohn-Sham orbitals  $\{\psi_i(\mathbf{r})\}$

$$E[\{\psi_i(\mathbf{r})\}] = -\frac{1}{2} \int \psi_i(\mathbf{r}) \nabla^2 \psi_i(\mathbf{r}) d\mathbf{r} + \int \rho(\mathbf{r}) V_{\text{ext}}(\mathbf{r}) d\mathbf{r} + \frac{1}{2} \int \frac{\rho(\mathbf{r})\rho(\mathbf{r}')}{|\mathbf{r} - \mathbf{r}'|} d\mathbf{r} d\mathbf{r}' + E_{\text{xc}}[\rho(\mathbf{r})] \quad (1.1)$$

where single-particle electron density  $\rho(\mathbf{r})$  is calculated as

$$\rho(\mathbf{r}) = \sum_i f_i \psi_i^*(\mathbf{r}) \psi_i(\mathbf{r}) \quad (1.2)$$

In Equation 1.2 the summation is done over all occupied single electronic states  $i$  and  $f_i$  is the occupation number of the  $i$ -th orbital. For doubly-occupied orbitals we have  $f_i = 2$ . In the usual setting in material science and chemistry, the external potential is usually the potential due to the atomic nuclei (ions):

$$V_{\text{ext}}(\mathbf{r}) = \sum_I \frac{Z_I}{|\mathbf{r} - \mathbf{R}_I|} \quad (1.3)$$

and an additional term of energy, the nucleus-nucleus energy  $E_{\text{nn}}$ , is added to the total energy functional (1.1):

$$E_{\text{nn}} = \frac{1}{2} \sum_I \sum_J \frac{Z_I Z_J}{|R_I - R_J|} \quad (1.4)$$

The energy terms in total energy functionals are the kinetic, external, Hartree, and exchange-correlation (XC) energy, respectively. The functional form of the last term (i.e. the XC energy) in terms of electron density is not known and one must resort to an approximation. In this article we will use the local density approximation for the XC energy. Under this approximation, the XC energy can be written as:

$$E_{\text{xc}}[\rho(\mathbf{r})] = \int \rho(\mathbf{r}) \epsilon_{\text{xc}}[\rho(\mathbf{r})] d\mathbf{r} \quad (1.5)$$

where  $\epsilon_{\text{xc}}$  is the exchange-correlation energy per particle. There are many functional forms that have been devised for  $\epsilon_{\text{xc}}$  and they are usually named by the persons who proposed them. An explicit form of  $\epsilon_{\text{xc}}$  will be given later.

Many material properties can be derived from the minimum of the functional (1.1). This minimum energy is also called the *ground state energy*. This energy can be obtained by using direct minimization of the Kohn-Sham energy functional or by solving the Kohn-Sham equations:

$$\hat{H}_{\text{KS}} \psi_i(\mathbf{r}) = \epsilon_i \psi_i(\mathbf{r}) \quad (1.6)$$

where  $\epsilon_i$  are the Kohn-Sham orbital energies and the Kohn-Sham Hamiltonian  $\hat{H}_{\text{KS}}$  is defined as

$$\hat{H}_{\text{KS}} = -\frac{1}{2} \nabla^2 + V_{\text{ext}}(\mathbf{r}) + V_{\text{Ha}}(\mathbf{r}) + V_{\text{xc}}(\mathbf{r}) \quad (1.7)$$

From the solutions of the Kohn-Sham equations:  $\epsilon_i$  and  $\psi_i(\mathbf{r})$  we can calculate the corresponding minimum total energy from the functional (1.1).

In the definition of Kohn-Sham Hamiltonian, other than the external potential which is usually specified from the problem, we have two additional potential terms, namely the Hartree and exchange-correlation potential. The Hartree potential can be calculated from its integral form

$$V_{\text{Ha}}(\mathbf{r}) = \int \frac{\rho(\mathbf{r}')}{|\mathbf{r} - \mathbf{r}'|} d\mathbf{r}' \quad (1.8)$$

An alternative way to calculate the Hartree potential is to solve the Poisson equation:

$$\nabla^2 V_{\text{Ha}}(\mathbf{r}) = -4\pi\rho(\mathbf{r}) \quad (1.9)$$

The exchange-correlation potential is defined as functional derivative of the exchange-correlation energy:

$$V_{\text{xc}}(\mathbf{r}) = \frac{\delta E[\rho(\mathbf{r})]}{\delta \rho(\mathbf{r})} \quad (1.10)$$

The Kohn-Sham equations are nonlinear eigenvalue equations in the sense that to calculate the solutions  $\{\epsilon_i\}$



and  $\{\psi_i(\mathbf{r})\}$  we need to know the electron density  $\rho(\mathbf{r})$  to build the Hamiltonian. The electron density itself must be calculated from  $\{\psi_i(\mathbf{r})\}$  which are not known (the quantities we want to solve for). The usual algorithm to solve the Kohn-Sham equations is the following:

- **(STEP 1)**: start from a guess input density  $\rho^{\text{in}}(\mathbf{r})$
- **(STEP 2)**: calculate the Hamiltonian defined in (1.7)
- **(STEP 3)**: solve the eigenvalue equations in (1.6) to obtain the eigenpairs  $\{\epsilon_i\}, \{\psi_i(\mathbf{r})\}$
- **(STEP 4)**: calculate the output electron density  $\rho^{\text{out}}(\mathbf{r})$  from  $\{\psi_i(\mathbf{r})\}$  that are obtained from the previous step.
- **(STEP 5)** Check whether the difference between  $\rho^{\text{in}}(\mathbf{r})$  and  $\rho^{\text{out}}(\mathbf{r})$  is small. If the difference is still above a chosen threshold then back to **STEP 1**. If the difference is already small the stop the algorithm.

The algorithm we have just described is known as the self-consistent field (SCF) algorithm.

Nowadays, there are many available software packages that can be used to solve the Kohn-Sham equations such as Quantum ESPRESSO, ABINIT, VASP, Gaussian, and NWChem are among the popular ones. A more complete list is given in a Wikipedia page [7]. These packages differs in several aspects, particularly the basis set used to discretize the Kohn-Sham equations and characteristic of the systems they can handle (periodic or non-periodic systems). These packages provide an easy way for researchers to carry out calculations based on density functional theory for particular systems they are interested in without knowing the details of how these calculations are performed.

In this book we will describe a simple way to solve the Kohn-Sham equations based on finite difference approximation. Our focus will be on the practical numerical implementation of a solver for the Kohn-Sham equations. The solver will be implemented using Julia programming language [8]. Our target is to calculate the ground state energy of several simple model systems. We will begin to build our solver starting from the ground up. The roadmap of the article is as follows.

- We begin from discussing numerical solution of Schroedinger equation in 1d. We will introduce finite difference approximation and its use in approximating second derivative operator that is present in the Schroedinger equation. We show how one can build the Hamiltonian matrix and solve the resulting eigenvalue equations using standard function that is available in Julia.
- In the next section, we discuss numerical solution of Schroedinger in 2d. We will introduce how one can handle 2d grid and applying finite difference approximation to the Laplacian operator present in the 2d Schroedinger equation. We also present several iterative methods to solve the eigenvalue equations.
- The next section discusses the numerical solution of Schroedinger equation in 3d. The methods presented in this section is a straightforward extension from the 2d case. In this section we start considering  $V_{\text{ext}}$  that is originated from the Coulomb interaction between atomic nuclei and electrons. We also introduce the concept of pseudopotential which is useful in practical calculations.
- The next section discusses about Poisson equation. Conjugate gradient method for solving system of linear equations. Hartree energy calculation.
- Hartree approximation, implementation of SCF algorithm
- Kohn-Sham equations, XC energy and potential, hydrogen molecule



## Chapter 2

# Schroedinger equation in 1d

In this chapter we will be concentrating on the problem of finding bound states solution to time-independent Schroedinger equation in one dimension:

$$\left[ -\frac{1}{2} \frac{d^2}{dx^2} + V(x) \right] \psi(x) = E \psi(x) \quad (2.1)$$

with the boundary conditions:

$$\lim_{x \rightarrow \pm\infty} \psi(x) = 0 \quad (2.2)$$

This boundary condition is relevant for non-periodic systems such as isolated or free atoms and molecules.

## 2.1 Grid points

We need to define a spatial domain  $[x_{\min}, x_{\max}]$  where  $x_{\min}, x_{\max}$  chosen such that the boundary condition 2.2 is approximately satisfied. The next step is to divide the spatial domain  $x$  using equally-spaced grid points which we will denote as  $\{x_1, x_2, \dots, x_N\}$  where  $N$  is total number of grid points. Various spatial quantities such as wave function  $\psi(x)$  and potential  $V(x)$  will be discretized on these grid points.

The grid points  $x_i, i = 1, 2, \dots$  are chosen to be:

$$x_i = x_{\min} + (i - 1)h \quad (2.3)$$

where  $h$  is the spacing between the grid points:

$$h = \frac{x_{\max} - x_{\min}}{N - 1} \quad (2.4)$$

The following Julia function can be used to initialize the grid points.

```
function init_FD1d_grid( x_min::Float64, x_max::Float64, N::Int64 )
    L = x_max - x_min
    h = L/(N-1) # spacing
    x = zeros(Float64,N) # the grid points
    for i = 1:N
        x[i] = x_min + (i-1)*h
    end
    return x, h
end
```

The function `init_FD1d_grid` takes three arguments:

- `x_min::Int64`: the left boundary point
- `x_max::Int64`: the right boundary point
- `N::Float64`: number of grid points

The function will return `x` which is an array of grid points and `h` which is the uniform spacing between grid points. The boundary points `x_min` and `x_max` will be included in the grid points.

As an example of the usage of the function `init_FD1d_grid`, let's sample and plot a Gaussian function

$$\psi(x) = e^{-\alpha x^2} \quad (2.5)$$

where  $\alpha$  is a positive number. We will sample the function within the domain  $[x_{\min}, x_{\max}]$  where  $x_{\min} = -5$  and  $x_{\max} = 5$ . The Gaussian function defined in (2.5) can be implemented as the following function.

```
function my_gaussian(x::Float64; α=1.0)
    return exp( -α*x^2 )
end
```

Note that we have set the default value of parameter  $\alpha$  to 1.

The full Julia program is as follows.

```
using Printf
using LaTeXStrings

import PyPlot
const plt = PyPlot
plt.rc("text", usetex=true)

include("init_FD1d_grid.jl")

function my_gaussian(x::Float64; α=1.0)
    return exp( -α*x^2 )
end

function main()
    A = -5.0
    B = 5.0
    Npoints = 8
    x, h = init_FD1d_grid( A, B, Npoints )
    @printf("Grid spacing = %f\n", h)
    @printf("\nGrid points:\n")
    for i in 1:Npoints
        @printf("%3d %18.10f\n", i, x[i])
    end
    NptsPlot = 200
    x_dense = range(A, stop=B, length=NptsPlot)
    plt.clf()
    plt.plot(x_dense, my_gaussian.(x_dense), label=L"f(x)")
    plt.plot(x, my_gaussian.(x), label=L"Sampled $f(x)$", marker="o")
    plt.legend()
    plt.tight_layout()
    plt.savefig("IMG_gaussian_1d_8pt.pdf")
end
```

```
main()
```

After execution, the program will print grid spacing and grid points to the standard output and also plot the function to a file named `IMG_gaussian_1d_8pt.pdf`. You may try to experiment by changing the value of  $N$  and compare the result. The resulting plots for  $N=8$  and  $N=21$  are shown in Figure XXX.




Figure 2.1: Sampling a Gaussian function defined in (2.5),  $\alpha = 1$ , two different number of grid points:  $N = 8$  (upper) and  $N = 21$  (lower). The sampled coordinates are marked by dots. The true or "continuous" function is emulated by using a dense sampling points of 200.

Note that we have used a densely-sampled points of `NptsPlot=200` in the program to emulate the true or "continuous" function. The plot of densely-sampled array is not done by not showing the point marker, as opposed to the array with lower sampling points.

**Exercise** Try to vary the value of  $\alpha$  and  $N$ . Make the program more sophisticated by using loop over for various values of  $N$  instead of manually changing its value in the program. You also may want to make set the saved filename of the resulting plot programmatically.

## 2.2 Approximating second derivative operator

Our next task is to find an approximation to the second derivative operator present in the Equation (2.1). Suppose that we have a function sampled at appropriate positions  $x_i$  as  $\psi(x_i)$ . How can we approximate  $\psi''(x_i)$ ? One simple approximation that we can use is the 3-point (central) finite difference:

$$\frac{d^2}{dx^2}\psi_i = \frac{\psi_{i+1} - 2\psi_i + \psi_{i-1}}{h^2} \quad (2.6)$$

where we have the following notation have been used:  $\psi_i = \psi(x_i)$ . Let's see we can apply this by writing out the Equation (2.6).

$$\begin{aligned} \psi_1'' &\approx (\psi_2 - 2\psi_1 + \psi_0) / h^2 \\ \psi_2'' &\approx (\psi_3 - 2\psi_2 + \psi_1) / h^2 \\ \psi_3'' &\approx (\psi_4 - 2\psi_3 + \psi_2) / h^2 \\ &\vdots \\ \psi_N'' &\approx (\psi_{N+1} - 2\psi_N + \psi_{N-1}) / h^2 \end{aligned}$$

In the first and the last equations, there are terms involving  $\psi_0$  and  $\psi_{N+1}$  which are not known. Recall that we have numbered our grid from 1 to  $N$ , so  $\psi_0$  and  $\psi_{N+1}$  are outside of our grid. However, by using the boundary equation (2.2), these quantities can be taken as zeros. So we have:

$$\begin{aligned} \psi_1'' &\approx (\psi_2 - 2\psi_1) / h^2 \\ \psi_2'' &\approx (\psi_3 - 2\psi_2 + \psi_1) / h^2 \\ \psi_3'' &\approx (\psi_4 - 2\psi_3 + \psi_2) / h^2 \\ &\vdots \\ \psi_N'' &\approx (-2\psi_N + \psi_{N-1}) / h^2 \end{aligned}$$

This operation can be compactly expressed by using matrix-vector notation. By taking  $\{\psi_i\}$  as a column vector, the second derivative operation can be expressed as matrix multiplication:

$$\{\psi''\} \approx \mathbb{D}^{(2)} \{\psi\} \quad (2.7)$$

where  $\mathbb{D}^{(2)}$  is the second derivative matrix operator:

$$\mathbb{D}^{(2)} = \frac{1}{h^2} \begin{bmatrix} -2 & 1 & 0 & 0 & 0 & \cdots & 0 \\ 1 & -2 & 1 & 0 & 0 & \cdots & 0 \\ 0 & 1 & -2 & 1 & 0 & \cdots & 0 \\ \vdots & \ddots & \ddots & \ddots & \ddots & \ddots & \vdots \\ 0 & \cdots & 0 & 1 & -2 & 1 & 0 \\ 0 & \cdots & \cdots & 0 & 1 & -2 & 1 \\ 0 & \cdots & \cdots & \cdots & 0 & 1 & -2 \end{bmatrix} \quad (2.8)$$

The following Julia function can be used to initialize the matrix  $\mathbb{D}^{(2)}$ .

```

function build_D2_matrix_3pt( N::Int64, h::Float64 )
    mat = zeros(Float64,N,N)
    for i = 1:N-1
        mat[i,i] = -2.0
        mat[i,i+1] = 1.0
        mat[i+1,i] = mat[i,i+1]
    end
    mat[N,N] = -2.0
    return mat/h^2
end

```

The function `build_D2_matrix_3pt` takes two arguments:

- `N::Int64`: number of grid points
- `h::Float64`: the uniform grid spacing

and returns the matrix  $\mathbb{D}^{(2)}$  as two dimensional array.

Before use these functions to solve Schroedinger equation, we will test the operation in Equation (2.8) for a simple function for which the second derivative can be calculated analytically. This function also should satisfy the boundary condition 2.2. We will take the the Gaussian function (2.5) that we have used before. The second derivative of this Gaussian function can be calculated as

$$\psi''(x) = (-2\alpha + 4\alpha^2 x^2) e^{-\alpha x^2} \quad (2.9)$$

We also need to define the computational domain  $[x_{\min}, x_{\max}]$  for our test. Let's choose  $x_{\min} = -5$  and  $x_{\max} = 5$  again as in the previous example. We can evaluate the value of function  $\psi(x)$  at those points to be at the order of  $10^{-11}$ , which is sufficiently small for our purpose.

The full Julia program that we will use is as follows.

```

using Printf
using LaTeXStrings

import PyPlot
const plt = PyPlot
plt.rc("text", usetex=true)

include("init_FD1d_grid.jl")
include("build_D2_matrix_3pt.jl")

function my_gaussian(x, α=1.0)
    return exp(-α*x^2)
end

function d2_my_gaussian(x, α=1.0)
    return (-2*α + 4*α^2 * x^2) * exp(-α*x^2)
end

function main(N::Int64)
    x_min = -5.0
    x_max = 5.0
    x, h = init_FD1d_grid( x_min, x_max, N )
    fx = my_gaussian.(x)

```

```

Ndense = 200
x_dense = range(A, stop=B, length=Ndense)
fx_dense = my_gaussian.(x_dense)
d2_fx_dense = d2_my_gaussian.(x_dense)

D2 = build_D2_matrix_3pt(N, h)
d2_fx_3pt = D2*fx

plt.clf()
plt.plot(x, fx, marker="o", label=L"Sampled $f(x)$")
plt.plot(x_dense, fx_dense, label=L"$f(x)$")
plt.plot(x, d2_fx_3pt, marker="o", label=L"Approx $f''(x)$")
plt.plot(x_dense, d2_fx_dense, label=L"$f''(x)$")
plt.legend()
plt.grid()
plt.savefig("IMG_gaussian_"*string(N)*".pdf")
end
main(15)
main(51)

```

We have followed similar approach as we have done in the previous section for plotting. The important parts of the program are the lines:

```

D2 = build_D2_matrix_3pt(N, h)
d2_fx_3pt = D2*fx

```

where we build  $\mathbb{D}^{(2)}$  matrix, represented by the variable D2 in the program, and multiply it with the vector fx to obtain an approximation to  $\psi''(x)$ . The results are plotted with two different values of number of grid points,  $N=15$  and  $N=51$ , which are shown in Figure XXX.




Figure 2.2: Finite difference approximation to a Gaussian function and its second derivative with number of grid points  $N = 15$






Figure 2.3: Finite difference approximation to a Gaussian function and its second derivative with number of grid points  $N = 51$

## 2.3 Harmonic potential

The time-independent Schroedinger equation (2.1) can be expressed into the following eigenvalue problem in matrix form:

$$\mathbb{H}\{\psi\} = E\{\psi\} \quad (2.10)$$

where  $\mathbb{H}$  is the Hamiltonian matrix and  $\{\psi\}$  is a vector column representation of wave function. In finite difference representation, the Hamiltonian matrix is

$$\mathbb{H} = -\frac{1}{2}\mathbb{D}^{(2)} + \mathbb{V} \quad (2.11)$$

where  $\mathbb{V}$  is a diagonal matrix whose elements are:

$$\mathbb{V}_{ij} = V(x_i)\delta_{ij} \quad (2.12)$$

As an example, we will start with a simple potential with known exact solution, namely the harmonic potential:

$$V(x) = \frac{1}{2}\omega^2 x^2 \quad (2.13)$$

In Julia, we can the process of building the Hamiltonian matrix is very simple

```
x, h = init_FD1d_grid(xmin, xmax, N) # grid points
D2 = build_D2_matrix_3pt(N, h)       # Build 2nd derivative matrix
Vpot = pot_harmonic.(x)              # Potential
Ham = -0.5*D2 + diagm( 0 => Vpot )  # Hamiltonian matrix
```

The function `pot_harmonic` is a function that calculate the harmonic potential

```
function pot_harmonic( x; ω=1.0 )
    return 0.5 * ω^2 * x^2
end
```

Once the Hamiltonian matrix is built, we can solve for  $E$  and  $\{\psi\}$  by using standard methods. In Julia this can be achieved by simply using the eigen function from LinearAlgebra standard library as illustrated in the following code.

```
evals, evecs = eigen( Ham )
```

The values of  $E$  or eigenvalues are stored in one-dimensional array `evals` and the corresponding wave functions or eigenvectors are stored by column in two-dimensional array or matrix `evecs`.

The complete Julia program to solve the Schroedinger equation for harmonic potential is as follows.

```
using Printf
using LinearAlgebra
using LaTeXStrings

import PyPlot
const plt = PyPlot
plt.rc("text", usetex=true)

include("init_FD1d_grid.jl")
include("build_D2_matrix_3pt.jl")

function pot_harmonic( x; ω=1.0 )
    return 0.5 * ω^2 * x^2
end

function main()
    xmin = -5.0
    xmax = 5.0
    N = 51
    x, h = init_FD1d_grid(xmin, xmax, N)
    D2 = build_D2_matrix_3pt(N, h)
    Vpot = pot_harmonic.(x)
    Ham = -0.5*D2 + diagm( 0 => Vpot )
    evals, evecs = eigen( Ham )
    # We will show the 5 lowest eigenvalues
    Nstates = 5
    @printf("Eigenvalues\n")
    ω = 1.0
    hbar = 1.0
    @printf(" State          Approx          Exact          Difference\n")
    for i in 1:Nstates
        E_ana = (2*i - 1)*ω*hbar/2
        @printf("%5d %18.10f %18.10f %18.10e\n", i, evals[i], E_ana, abs(evals[i]-E_ana))
    end

    # normalize the first three eigenstates
    for i in 1:3
        ss = dot(evecs[:,i], evecs[:,i])*h
        evecs[:,i] = evecs[:,i]/sqrt(ss)
    end
end
```

```

# Plot up to 3rd eigenstate
plot_title = "N="*string(N)
plt.plot(x, evecs[:,1], label="1st eigenstate", marker="o")
plt.plot(x, evecs[:,2], label="2nd eigenstate", marker="o")
plt.plot(x, evecs[:,3], label="3rd eigenstate", marker="o")
plt.legend()
plt.grid()
plt.tight_layout()
plt.savefig("IMG_main_harmonic_01_"*string(N)*".pdf")
end

main()

```

The program shows five lowest approximate eigenvalues and their comparison with analytical or exact eigenvalues. A sample output from running the program is as follows.

| Eigenvalues |              |              |                  |
|-------------|--------------|--------------|------------------|
| State       | Approx       | Exact        | Difference       |
| 1           | 0.4987468513 | 0.5000000000 | 1.2531486828e-03 |
| 2           | 1.4937215179 | 1.5000000000 | 6.2784821079e-03 |
| 3           | 2.4836386480 | 2.5000000000 | 1.6361352013e-02 |
| 4           | 3.4684589732 | 3.5000000000 | 3.1541026791e-02 |
| 5           | 4.4481438504 | 4.5000000000 | 5.1856149551e-02 |

The program will also plot three lowest resulting eigenvectors (wave functions) to a file. An example plot is shown in Figure 2.4.




Figure 2.4: Eigenstates of harmonic oscillator

You can get more accurate values of eigenvalues by using higher value for grid points. This is the result if we use  $N = 81$ .

| Eigenvalues |              |              |                  |
|-------------|--------------|--------------|------------------|
| State       | Approx       | Exact        | Difference       |
| 1           | 0.4995112405 | 0.5000000000 | 4.8875946328e-04 |
| 2           | 1.4975542824 | 1.5000000000 | 2.4457175928e-03 |

|   |              |              |                  |
|---|--------------|--------------|------------------|
| 3 | 2.4936355672 | 2.5000000000 | 6.3644328157e-03 |
| 4 | 3.4877496130 | 3.5000000000 | 1.2250387024e-02 |
| 5 | 4.4798939398 | 4.5000000000 | 2.0106060206e-02 |

## 2.4 Higher order finite difference

To obtain higher accuracy, we can include more points in our approximation to second derivative operator. An example is by using 5-point finite-difference formula:

$$\frac{d^2}{dx^2}\psi_i = \frac{-\psi_{i+2} + 16\psi_{i+1} - 30\psi_i + 16\psi_{i-1} - \psi_{i-2}}{12h^2} \quad (2.14)$$

The corresponding second derivative matrix can be built using the following function.

```
function build_D2_matrix_5pt( N::Int64, h::Float64 )
    mat = zeros(Float64,N,N)
    for i = 1:N-2
        mat[i,i] = -30.0
        mat[i,i+1] = 16.0
        mat[i,i+2] = -1.0
        mat[i+1,i] = mat[i,i+1]
        mat[i+2,i] = mat[i,i+2]
    end
    #
    mat[N-1,N-1] = -30.0
    mat[N-1,N] = 16.0
    mat[N,N-1] = mat[N-2,N-1]
    mat[N,N] = -30.0
    #
    return mat/(12*h^2)
end
```

We will refer to the number of points used to approximate the derivative operator as stencil order, thus the Equation (2.14) has stencil order of 5 and Equation (2.6) has stencil order of 3. Other approximation formulas for second derivative also can be used. In the source code repository, we have implemented 3-, 5-, 7-, 9-, and 11-points formulas. The web page <http://web.media.mit.edu/~crtaylor/calculator.html> can be consulted to obtain the coefficients of finite-difference approximation.

In Figure 2.5 we plot the difference (approximation - exact) of second derivative of the Gaussian function for several approximation formulas using  $N = 15$  grid points. The errors have different magnitude for each points with the maximum error occurs at the center of the Gaussian. We can readily see that the errors for each point are decreasing with increasing stencil order.

## 2.5 Harmonic potential with higher order finite-difference formulas

We can compare the effect of different stencil order to the accuracy of numerical approximation to the Schroedinger equation. In the following program, we try to compare the accuracy of first eigenvalue of Schroedinger equation with harmonic potential.




Figure 2.5: Error of 2nd derivative approx

```

function solve_eigvals(N::Int64, D2_func)
    # Initialize the grid points
    x_min = -5.0
    x_max = 5.0
    x, h = init_FD1d_grid(x_min, x_max, N)
    # Build 2nd derivative matrix
    D2 = D2_func(N, h)
    # Potential
    Vpot = pot_harmonic.(x)
    # Hamiltonian
    Ham = -0.5*D2 + diagm( 0 => Vpot )
    # Solve for the eigenvalues only
    evals = eigvals( Ham )
    #
    return evals
end

function main()
    Npoints = 20
    funcs = [build_D2_matrix_3pt, build_D2_matrix_5pt, build_D2_matrix_7pt,
             build_D2_matrix_9pt, build_D2_matrix_11pt]

    ω = 1.0
    hbar = 1.0
    ist = 1
    E_ana = (2*ist - 1)*ω*hbar/2

    for f in funcs
        evals = solve_eigvals(Npoints, f)
        dE = abs(evals[ist]-E_ana) # absolute error
        @printf("%20s: %18.10f %18.10e\n", string(f), evals[ist], dE)
    end
end

```

```
end
```

```
main()
```

The results for fixed point ( $N = 20$ ) and different stencil order are given below. The second column is the first eigenvalue and the third column is the difference with exact value (0.5 Hartree). The results clearly shows that the error decreases with increasing stencil order.

|                       |              |                  |
|-----------------------|--------------|------------------|
| build_D2_matrix_3pt:  | 0.4911851752 | 8.8148248058e-03 |
| build_D2_matrix_5pt:  | 0.4992583463 | 7.4165366457e-04 |
| build_D2_matrix_7pt:  | 0.4998965249 | 1.0347512122e-04 |
| build_D2_matrix_9pt:  | 0.4999802170 | 1.9782954165e-05 |
| build_D2_matrix_11pt: | 0.4999952673 | 4.7326636594e-06 |

## 2.6 Exercises

Potential well

Gaussian potential

Periodic boundary condition

## Chapter 3

# Schroedinger equation in 2d

Now we will turn our attention to higher dimensions, i.e 2d. The time-independent Schrodinger equation in 2d can be written as:

$$\left[ -\frac{1}{2}\nabla^2 + V(x, y) \right] \psi(x, y) = E \psi(x, y) \quad (3.1)$$

where  $\nabla^2$  is the Laplacian operator:

$$\nabla^2 = \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} \quad (3.2)$$

### 3.1 Describing grid in 2d

Now we have two directions  $x$  and  $y$ . Our approach to solving the Schroedinger equation is similar to the one we have used before in 1d, however several technical difficulties will arise.

To describe the computational grid, we now need to specify  $x_{\max}, x_{\min}$  for the  $x$ -domain and  $y_{\max}, y_{\min}$  for  $y$ -domain. We also need to specify number of grid points in for each  $x$  and  $y$ -directions, i.e.  $N_x$  and  $N_y$ . There are quite lot of variables. For easier management, we will collect our grid related variables in one data structure or struct in Julia. A struct in Julia looks very much like C-struct. It also defines a new custom data type in Julia.

Our struct definition looks like this.

```
struct FD2dGrid
    Npoints::Int64
    Lx::Float64
    Ly::Float64
    Nx::Int64
    Ny::Int64
    hx::Float64
    hy::Float64
    dA::Float64
    x::Array{Float64,1}
    y::Array{Float64,1}
    r::Array{Float64,2}
    idx_ip2xy::Array{Int64,2}
    idx_xy2ip::Array{Int64,2}
    pbc::Tuple{Bool,Bool}
end
```

An instance of FD2dGrid can be initialized using the following constructor function:

```

function FD2dGrid( x_domain, Nx, y_domain, Ny )
    x, hx = init_FD1d_grid(x_domain, Nx)
    y, hy = init_FD1d_grid(y_domain, Ny)
    dA = hx*hy
    Npoints = Nx*Ny
    r = zeros(2,Npoints)
    ip = 0
    idx_ip2xy = zeros(Int64,2,Npoints)
    idx_xy2ip = zeros(Int64,Nx,Ny)
    for j in 1:Ny
        for i in 1:Nx
            ip = ip + 1
            r[1,ip] = x[i]
            r[2,ip] = y[j]
            idx_ip2xy[1,ip] = i
            idx_ip2xy[2,ip] = j
            idx_xy2ip[i,j] = ip
        end
    end
    return FD2dGrid(Npoints, Nx, Ny, hx, hy, dA, x, y, r, idx_ip2xy, idx_xy2ip)
end

```

A short explanation about the members of FD2dGrid follows.

- Npoints is the total number of grid points.
- Nx and Ny is the total number of grid points in  $x$  and  $y$ -directions, respectively.
- hx and hy is grid spacing in  $x$  and  $y$ -directions, respectively. dA is the product of hx and hy.
- x and y are the grid points in  $x$  and  $y$ -directions. The actual two dimensional grid points  $r \equiv (x_i, y_i)$  are stored as two dimensional array  $r$ .
- Thw two integers arrays idx\_ip2xy and idx\_xy2ip defines mapping between two dimensional grids and linear grids.

As an illustration let's build a grid for a rectangular domain  $x_{\min} = y_{\min} = -5$  and  $x_{\max} = y_{\max} = 5$  and  $N_x = 3$ ,  $N_y = 4$ . Using the above constructor for FD2dGrid:

```

Nx = 3
Ny = 4
grid = FD2dGrid( (-5.0,5.0), Nx, (-5.0,5.0), Ny )

```

Dividing the  $x$  and  $y$  accordingly we obtain  $N_x = 3$  grid points along  $x$ -direction

```

julia> println(grid.x)
[-5.0, 0.0, 5.0]

```

and  $N_y = 4$  points along the  $y$ -direction

```

julia> println(grid.y)
[-5.0, -1.6666666666666665, 1.6666666666666667, 5.0]

```

The actual grid points are stored in grid.r. Using the following snippet, we can printout all of the grid points:



```

for ip = 1:grid.Npoints
    @printf("%3d %8.3f %8.3f\n", ip, grid.r[1,ip], grid.r[2,ip])
end

```

The results are:

```

 1  -5.000  -5.000
 2   0.000  -5.000
 3   5.000  -5.000
 4  -5.000  -1.667
 5   0.000  -1.667
 6   5.000  -1.667
 7  -5.000   1.667
 8   0.000   1.667
 9   5.000   1.667
10  -5.000   5.000
11   0.000   5.000
12   5.000   5.000

```

We also can use the usual rearrange these points in the usual 2d grid rearrangement:

```

[ -5.000, -5.000] [ -5.000, -1.667] [ -5.000,  1.667] [ -5.000,  5.000]
[  0.000, -5.000] [  0.000, -1.667] [  0.000,  1.667] [  0.000,  5.000]
[  5.000, -5.000] [  5.000, -1.667] [  5.000,  1.667] [  5.000,  5.000]

```

which can be produced from the following snippet:

```

for i = 1:Nx
    for j = 1:Ny
        ip = grid.idx_xy2ip[i,j]
        @printf("[%8.3f,%8.3f] ", grid.r[1,ip], grid.r[2,ip])
    end
    @printf("\n")
end

```

## 3.2 Laplacian operator

Having built out 2d grid, we now turn our attention to the second derivative operator or the Laplacian in the equation 3.1. There are several ways to build a matrix representation of the Laplacian, but we will use the easiest one.

Before constructing the Laplacian matrix, there is an important observation that we should make about the second derivative matrix  $\mathbb{D}^{(2)}$ . We should note that the second derivative matrix contains mostly zeros. This type of matrix that most of its elements are zeros is called **sparse matrix**. In a sparse matrix data structure, we only store its non-zero elements with specific formats such as compressed sparse row/column format (CSR/CSC) and coordinate format. We have not made use of the sparsity of the second derivative matrix in the 1d case for simplicity. In the higher dimensions, however, we must make use of this sparsity, otherwise we will waste computational resources by storing many zeros. The Laplacian matrix that we will build from  $\mathbb{D}^{(2)}$  is also very sparse.

Given second derivative matrix in  $x$ ,  $\mathbb{D}_x^{(2)}$ ,  $y$  direction,  $\mathbb{D}_y^{(2)}$ , we can construct finite difference representation of the Laplacian operator  $\mathbb{L}$  by using

$$\mathbb{L} = \mathbb{D}_x^{(2)} \otimes \mathbb{I}_y + \mathbb{I}_x \otimes \mathbb{D}_y^{(2)} \quad (3.3)$$

where  $\otimes$  is Kronecker product. In Julia, we can use the function `kron` to form the Kronecker product between two matrices `A` and `B` as `kron(A,B)`.

The following function illustrates the above approach to construct matrix representation of the Laplacian operator.

```
function build_nabla2_matrix( grid::FD2dGrid )
    Nx = grid.Nx; hx = grid.hx
    Ny = grid.Ny; hy = grid.hy
    D2x = build_D2_matrix_9pt(Nx, hx)
    D2y = build_D2_matrix_9pt(Ny, hy)
    ∇2 = kron(D2x, speye(Ny)) + kron(speye(Nx), D2y)
    return ∇2
end
```

The standard Julia library does not include definition for `speye` function but it can be implemented by the following definition.

```
speye(N::Int64) = sparse( Matrix{Float64}(I, N, N) )
```

In the Figure 3.1, an example to the approximation of 2nd derivative of 2d Gaussian function by using finite difference is shown.




Figure 3.1: Two-dimensional Gaussian function and its finite difference approximation of second derivative

The following program is used to produce the figure.

```
using Printf
using LinearAlgebra
using SparseArrays

import PyPlot
const plt = PyPlot

include("FD2dGrid.jl")
include("build_nabla2_matrix.jl")
include("supporting_functions.jl")

function my_gaussian( grid; α=1.0 )
```

```

Npoints = grid.Npoints
f = zeros(Npoints)
for i in 1:Npoints
    x = grid.r[1,i]
    y = grid.r[2,i]
    r2 = x^2 + y^2
    f[i] = exp(-α*r2)
end
return f
end

function main()
    Nx = 75
    Ny = 75
    grid = FD2dGrid( (-5.0,5.0), Nx, (-5.0,5.0), Ny )

    ∇2 = build_nabla2_matrix( grid )

    fg = my_gaussian(grid, α=0.5)
    plt.clf()
    plt surf(grid.x, grid.y, reshape(fg, grid.Nx, grid.Ny), cmap=:jet)
    plt.gca(projection="3d").view_init(30,7)
    fileplot = "IMG_gaussian2d.pdf"
    plt.savefig(fileplot)

    d2fg = ∇2*fg
    plt.clf()
    plt surf(grid.x, grid.y, reshape(d2fg, grid.Nx, grid.Ny), cmap=:jet)
    plt.gca(projection="3d").view_init(30,7)
    fileplot = "IMG_d2_gaussian2d.pdf"
    plt.savefig(fileplot)
end

main()

```

### 3.3 More about sparse matrices in Julia

Using the function `typeof` we can know the type of the variable `∇2` which was returned by the function `build_nabla2_matrix`:

```

julia> typeof(∇2)
SparseMatrixCSC{Float64,Int64}

```

The CSC part here is actually describe a certain format to store a sparse matrix. It stands for **C**ompressed **S**parse **C**olumn format. There are other formats as well: such as Compressed Sparse Row format (CSR), coordinate list format (COO), and several others.

The CSC format stores a sparse matrix **A** with size  $m \times n$  using three (one-dimensional) arrays (`nzval`, `colptr`, `rowval`). Let `Nnz` denote the number of nonzero entries in **A**.

The arrays `nzval` and `rowval` are of length `Nnz`, and contain the non-zero values and the row indices of those values respectively.

An integer array `colptr` contains pointers to the beginning of each column in the arrays `nzval` and `rowval`. Thus the content of `colptr[i]` is the position in arrays `nzval` and `rowval` where the *i*-th row starts. The length of `colptr`

is  $n + 1$  with `colptr[n + 1]` containing the number `colptr[1] + Nnz`, i.e., the address in `nzval` and `rowval` of the beginning of a fictitious column  $m + 1$ .

The array `colptr` has one element per column in the matrix and encodes the index in `nzval` where the given column starts. It may contain an extra end element which is set to `Nnz`.

Field names:

```
julia> fieldnames(typeof(V2))
(:m, :n, :colptr, :rowval, :nzval)
```

From the documentation

```
struct SparseMatrixCSC{Tv,Ti<:Integer} <: AbstractSparseMatrix{Tv,Ti}
  m::Int          # Number of rows
  n::Int          # Number of columns
  colptr::Vector{Ti} # Column j is in colptr[j]:(colptr[j+1]-1)
  rowval::Vector{Ti} # Row indices of stored values
  nzval::Vector{Tv}  # Stored values, typically nonzeros
end
```

Example 1:

$$A = \begin{bmatrix} 0 & 0 & 1.0 & 0.0 \\ 5 & 8 & 0.0 & 0.0 \\ 0 & 0 & 3.0 & 0.0 \\ 0 & 6 & 0.0 & 0.0 \\ 0 & 0 & 0.0 & 0.0 \end{bmatrix} \quad (3.4)$$

```
nzval = [5.0, 8.0, 6.0, 1.0, 3.0]
rowval = [2, 2, 4, 1, 3]
colptr = [1, 2, 4, 6, 6]
```

Example 2:

$$\mathbf{B} = \begin{bmatrix} 0 & 0 & 1 & 5 \\ 5 & 0 & 0 & 0 \\ 0 & 0 & 3 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 5 & 0 & 0 & 6 \end{bmatrix} \quad (3.5)$$

```
nzval = [5.0, 5.0, 1.0, 3.0, 1.0, 5.0, 6.0]
rowval = [2, 6, 1, 3, 5, 1, 6]
colptr = [1, 3, 3, 6, 8]
```

Example 3:

$$\mathbf{C} = \begin{bmatrix} -2 & 1 & 0 & 0 & 0 \\ 1 & -2 & 1 & 0 & 0 \\ 0 & 1 & -2 & 1 & 0 \\ 0 & 0 & 1 & -2 & 1 \\ 0 & 0 & 0 & 1 & -2 \end{bmatrix} \quad (3.6)$$

```
nzval = [-2.0, 1.0, 1.0, -2.0, 1.0, 1.0, -2.0, 1.0, 1.0, -2.0, 1.0, 1.0, -2.0]
rowval = [1, 2, 1, 2, 3, 2, 3, 4, 3, 4, 5, 4, 5]
colptr = [1, 3, 6, 9, 12, 14]
```

Another way: `sparse(I,J,V)` then constructs a sparse matrix such that  $S[I[k], J[k]] = V[k]$ .

### 3.4 Iterative methods for eigenvalue problem

Now that we know how to build the Laplacian matrix, we now can build the Hamiltonian matrix given some potential:

```
∇2 = build_nabla2_matrix( grid )
Ham = -0.5*∇2 + spdiags( 0 => Vpot )
```

Note that we have used sparse diagonal matrix for building the potential matrix by using the function `spdiags`. Our next task after building the Hamiltonian matrix is to find the eigenvalues and eigenfunctions. However, note that the Hamiltonian matrix size is large. For example, if we use  $N_x = 50$  and  $N_y = 50$  we will end up with a Hamiltonian matrix with the size of 2500. The use of eigen method, as we have done in the 1d case, to solve this eigenvalue problem is thus not practical. Actually, given enough computer memory and time, we can use the function `eigen` anyway to find all eigenvalue and eigenfunction pairs of the Hamiltonian, however it is not recommended nor practical for larger problem size.

Typically, we do not need to solve for all eigenvalue and eigenfunction pairs. We only need to solve for several eigenpairs with lowest eigenvalues. In a typical density functional theory calculations, we only need to solve for  $N_{\text{electrons}}$  or  $N_{\text{electrons}}/2$  lowest states, where  $N_{\text{electrons}}$  is the number of electrons in the system.

In numerical methods, there are several methods to search for several eigenpairs of a matrix. These methods falls into the category of *partial or iterative diagonalization methods*. Several known methods are Lanczos method, Davidson method, preconditioned conjugate gradients, etc. More detailed discussion about these methods are deferred to Appendix XXX. We have prepared several implementation of iterative diagonalization methods which can be used for black boxes for your convenience:

- `diag_Emin_PCG`
- `diag_davidson`
- `diag_LOBPCG`

These functions have similar function signatures. They typically need Hamiltonian, initial guess of wave function and preconditioner. An example of `diag_LOBPCG` is given below.

```
function diag_LOBPCG!( Ham, X::Array{Float64,2}, prec;
    tol=1e-5, NiterMax=100, verbose=false,
    verbose_last=false, Nstates_conv=0 )
```

The three mandatory arguments are as follow:

- `Ham`: the Hamiltonian matrix
- `X::Array{Float64,2}`: initial guess of eigenfunctions
- `prec`: the preconditioner

Almost all iterative methods need a good preconditioner to function properly. We will use several ready-to-use preconditioners that have been implemented in several packages in Julia such as incomplete LU and multigrid preconditioners. In the next section we will describe a diagonalization method based on nonlinear minimization as we apply it to a simple case of 2d harmonic potential.

### 3.5 Minimization approach: 2d harmonic potential

We will test our implementation for solving Schroedinger equation for two dimensional harmonic potentials:

$$V(x, y) = \frac{1}{2}\omega^2(x^2 + y^2) \quad (3.7)$$

This potential can be implemented in the following Julia code.

```
function pot_harmonic( grid::FD2dGrid; ω=1.0 )
    Npoints = grid.Npoints
    Vpot = zeros(Npoints)
    for i in 1:Npoints
        x = grid.r[1,i]
        y = grid.r[2,i]
        Vpot[i] = 0.5 * ω^2 * ( x^2 + y^2 )
    end
    return Vpot
end
```

This potential has the following analytic solutions for eigenvalues:

$$E_{n_x+n_y} = \hbar\omega (n_x + n_y + 1) \quad (3.8)$$

For each energy levels, we may have more than one possible combinations of  $n_x$  and  $n_y$ , for examples:

| $E = n_x + n_y + 1$ | Values of $n_x$ and $n_y$ |
|---------------------|---------------------------|
| 1                   | (0,0)                     |
| 2                   | (1,0) (0,1)               |
| 3                   | (2,0) (1,1) (1,1)         |
| 4                   | (3,0) (0,3) (2,1) (1,2)   |

Our first task is to build the Hamiltonian. The step is very similar to the one that we have done for 1d case:

```
Nx = 50
Ny = 50
grid = FD2dGrid( (-5.0,5.0), Nx, (-5.0,5.0), Ny )
∇2 = build_nabla2_matrix( grid )
Vpot = pot_harmonic( grid )
Ham = -0.5*∇2 + spdiag( 0 => Vpot )
```

Note that we have used `spdiagm` instead of `diagm` when constructing the potential matrix.

#### 3.5.1 Orthonormalization

We usually need to provide a guess solution or vectors to our eigensolver. The guess vectors should be orthonormalized properly before they can be supplied to the eigensolver. There are several algorithms that can be used

to orthonormalize the vectors. One of them that will be used here is the Lowdin orthonormalization which can be implemented in Julia as:

```
function ortho_sqrt( psi )
    Udagger = inv(sqrt(psi'*psi))
    return psi*Udagger
end
```

As an example usage, here we start from random vectors and orthonormalize them with `ortho_sqrt`:

```
dVol = grid.dVol
Nstates = 3
Npoints = Nx*Ny
X = rand(Float64, Npoints, Nstates)
ortho_sqrt!(X, dVol)
```

TODO: Gram-Schmidt ortho

TODO: check that the vectors are properly orthonormalized.

### 3.5.2 Band energy functional and its gradient

One method that can be used to diagonalize the Hamiltonian is based on the minimization of band energy:

$$\min E[\psi_i] = \langle \psi | H | \psi \rangle - \sum_{ij} \lambda_{ij} (\langle \psi_i | \psi_j \rangle - \delta_{ij}) \quad (3.9)$$

The gradient of this expression is:

$$\frac{\partial H}{\partial \psi_i^*} = H\psi_i - \sum_j \psi_j \langle \psi_j | H | \psi_i \rangle \quad (3.10)$$

In Julia this can be implemented using the following function:

```
function calc_grad_evals!( Ham, ψ, g, Hsub )
    Nstates = size(ψ,2)
    Hψ = Ham*ψ
    Hsub[:] = ψ' * Hψ
    g[:, :] = Hψ - ψ*Hsub
    return
end
```

### 3.5.3 Steepest descent method

The simplest method that we can apply to minimize the band energy functional is the steepest descent method. This method can be described in the following steps:

1. Initialize random wave function: **X**
2. Calculate  $E$  for the given **X**.
3. Calculate the gradient  $\mathbf{g} = \nabla E$ .
4. Set search direction  $\mathbf{d} = -\mathbf{g}$ .

5. Update wave function according to:  $\mathbf{X} \leftarrow \mathbf{X} + \alpha d$ , where  $\alpha$  is a fixed step length. Orthonormalize  $\mathbf{X}$ .
6. Calculate  $E$  for this updated  $\mathbf{X}$  and compare it with the previous value.

The following Julia code implements the steepest descent algorithm for minimizing band energy.

```
for iter = 1:NiterMax
    calc_grad_evals!( Ham, X, g, Hsub )
    d[:] = -g[:]
    # Update wavefunction
    X[:] = X + α_t*d
    ortho_sqrt!(X)
    Hr = Hermitian( X' * ( Ham*X ) )
    evals = eigvals(Hr)
    Ebands = sum(evals)
    devals = abs.( evals - evals_old )
    evals_old = copy(evals)
    nconv = length( findall( devals .< tol ) )
    diffE = abs(Ebands-Ebands_old)
    @printf("SD step %8d = %18.10f    %10.7e  nconv = %5d\n", iter, Ebands, diffE, nconv)
    if nconv >= Nstates_conv
        IS_CONVERGED = true
        @printf("Emin_SD convergence: nconv = %5d in %5d iterations\n", nconv, iter)
        break
    end
    Ebands_old = Ebands
end
```

Note we have calculate the current estimate the eigenvalues and band energy in the following lines:

```
Hr = Hermitian( X' * ( Ham*X ) )
evals = eigvals(Hr)
Ebands = sum(evals)
```

In the file `main_harmonic_Emin_SD.kl` we implement steepest descent algorithm to find three lowest eigenstates of 2d harmonic potential. Using the following parameters:

```
NiterMax = 5000
α = 3e-3
tol = 1e-6
```

we obtain the results:

```
evals[ 1] =      0.9999999732  devals =  5.2402526762e-14
evals[ 2] =      2.0000187702  devals =  1.1408453116e-07
evals[ 3] =      2.0001656321  devals =  9.9907378681e-07
```

which are close to the analytical solution. You might want to vary the number of eigenstates that you want to search by setting higher values for the variable `Nstates` to higher than 3.

Despite its simplicity, the steepest descent method has several drawbacks. The most prominent one is that it is very slow to converge. For this particular case we have:



```

SD step      2258 =      5.0001854888  1.1198655e-06  nconv =    2
SD step      2259 =      5.0001843756  1.1131584e-06  nconv =    3
Emin_SD convergence: nconv =    3 in  2259 iterations

```

The convergence is also dependent to the step length parameter  $\alpha$ .

### 3.5.4 Line minimization

Line minimization:

```

Xc = ortho_sqrt( X +  $\alpha_t$ *d )
calc_grad_evals!( Ham, Xc, gt, Hsub )
denum = real(sum(conj(g-gt).*d))
if denum != 0.0
     $\alpha$  = abs(  $\alpha_t$ *real(sum(conj(g).*d))/denum )
else
     $\alpha$  = 0.0
end

```

Implemented in the program main\_harmonic\_Emin\_linmin.jl. Result using line minimization:

```

linmin step    589 =      5.0000566204  1.4525395e-06  nconv =    2
linmin step    590 =      5.0000552040  1.4164238e-06  nconv =    3
Emin_linmin convergence: nconv =    3 in  590 iterations

```

Eigenvalues:

```

evals[ 1] =      1.0000001750  devals =  3.4402208904e-07
evals[ 2] =      2.0000051631  devals =  1.0311948540e-07
evals[ 3] =      2.0000498659  devals =  9.6928218651e-07

```

### 3.5.5 Preconditioning

Action of preconditioning:

```

Kg[:] = g[:] # copy
for i in 1:Nstates
    @views ldiv!(prec, Kg[:,i])
end

```

ILU preconditioner based on kinetic operator:

```
prec = ilu(-0.5* $\nabla^2$ )
```

Result using ILU preconditioner based on kinetic operator:

```

linmin step    55 =      5.0000009960  1.2258080e-06  nconv =    2
linmin step    56 =      5.0000003689  6.2711750e-07  nconv =    3
Emin_linmin convergence: nconv =    3 in  56 iterations

```

Eigenvalues:

```

evals[ 1] =      0.9999999739  devals =  6.0261565960e-07
evals[ 2] =      1.9999999937  devals =  5.8729066055e-09

```

```
evals[ 3] =      2.0000004014 devals = 1.8628935727e-08
```

ILU preconditioner based on Hamiltonian operator:

```
prec = ilu(Ham)
```

```
linmin step      15 =      5.0000004295  3.2233493e-06  nconv =      2
linmin step      16 =      4.9999998058  6.2374058e-07  nconv =      3
Emin_linmin convergence: nconv =      3 in      16 iterations
```

Eigenvalues:

```
evals[ 1] =      0.9999999787 devals = 1.2492943446e-07
evals[ 2] =      1.9999998478 devals = 1.2682084205e-07
evals[ 3] =      1.9999999792 devals = 3.7199030434e-07
```

Multigrid preconditioner

```
prec = aspreconditioner(ruge_stuben(Ham))
```

```
linmin step      15 =      5.0000010097  4.8543236e-06  nconv =      2
linmin step      16 =      4.9999999515  1.0581623e-06  nconv =      3
Emin_linmin convergence: nconv =      3 in      16 iterations
```

Eigenvalues:

```
evals[ 1] =      0.9999999796 devals = 1.0151917307e-07
evals[ 2] =      1.9999998532 devals = 1.9803365658e-07
evals[ 3] =      2.0000001187 devals = 7.5860946458e-07
```

ILU0 preconditioner, in SPARSKIT.

```
prec = ILU0Preconditioner(Ham)
```

```
linmin step      64 =      5.0000041335  1.4349680e-06  nconv =      2
linmin step      65 =      5.0000030469  1.0865516e-06  nconv =      3
Emin_linmin convergence: nconv =      3 in      65 iterations
```

Eigenvalues:

```
evals[ 1] =      1.0000001632 devals = 1.3899380646e-07
evals[ 2] =      2.0000000485 devals = 6.1344835878e-08
evals[ 3] =      2.0000028351 devals = 8.8621293415e-07
```

For testing purpose

```
prec = NoPreconditioner()
```

Comparison of preconditioner size

```

sizeof Ham =      0.6372146606 MiB
sizeof prec =     6.9084167480 MiB  ILU kinetic
sizeof prec =     3.0494079590 MiB  ILU Hamiltonian
sizeof prec =     2.3046264648 MiB  AMG Ruge-Stuben
sizeof prec =     0.6372070313 MiB  ILU0

```

### 3.5.6 Conjugate gradient

Conjugate-gradient

```
d = -Kg +  $\beta$  * d_prev
```

Polak-Ribiere formula

```

if iter != 1
     $\beta$  = real(sum(conj(g-g_prev).*Kg))/real(sum(conj(g_prev).*Kg_prev))
end
if  $\beta$  < 0.0  $\beta$  = 0.0 end

```

CG using ILU0 (Ham)

```

CG step      29 =      5.00000026056    2.8129344e-06  nconv =      2
CG step      30 =      5.00000011500    1.4555620e-06  nconv =      3
Emin_CG convergence: nconv =      3 in      30 iterations

```

Eigenvalues:

```

evals[ 1] =      1.00000001156  devals =      1.4606434040e-07
evals[ 2] =      2.00000002046  devals =      6.6796002196e-07
evals[ 3] =      2.00000008298  devals =      6.4153765811e-07

```

### 3.5.7 Eigenfunctions

```

ortho_sqrt!(X)
Hr = Hermitian( X' * (Ham*X) )
evals, evecs = eigen(Hr)
X[:, :] = X*evecs

```

The eigenfunctions are shown in Figure 3.2.




Figure 3.2: Visualization of eigenstates of 2d harmonic potential

## 3.6 Exercises

2d Gaussian potential

## Chapter 4

# Schroedinger equation in 3d

After we have considered two-dimensional Schroedinger equations, we are now ready for the extension to three-dimensional systems. In 3d, Schroedinger equation can be written as:

$$\left[ -\frac{1}{2}\nabla^2 + V(\mathbf{r}) \right] \psi(\mathbf{r}) = E \psi(\mathbf{r}) \quad (4.1)$$

where  $\mathbf{r}$  is the abbreviation to  $(x, y, z)$  and  $\nabla^2$  is the Laplacian operator in 3d:

$$\nabla^2 = \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} + \frac{\partial^2}{\partial z^2} \quad (4.2)$$

### 4.0.1 Three-dimensional grid

As in the preceeding chapter, our first task is to create a representation of 3d grid points and various quantities defined on it. This task is realized using straightforward extension of FD2dGrid to FD3dGrid.

Visualization of 3d functions as isosurface map or slice of 3d array.

Introducing 3d xsf

### 4.0.2 Laplacian operator

The Laplacian in 3d can be written as:

$$\mathbb{L} = \mathbb{D}_x^{(2)} \otimes \mathbb{I}_y \otimes \mathbb{I}_z + \mathbb{I}_x \otimes \mathbb{D}_y^{(2)} \otimes \mathbb{I}_z + \mathbb{I}_x \otimes \mathbb{I}_y \otimes \mathbb{D}_z^{(2)} \quad (4.3)$$

The Julia code is also similar to the one we have used for the case of 2d:

```
const ⊗ = kron
function build_nabla2_matrix( grid )
    # ... SNIPPED
    # build D2x, D2y, and D2z matrices
    IIX = speye(grid.Nx)
    IIY = speye(grid.Ny)
    IIZ = speye(grid.Nz)
    ∇2 = D2x⊗IIY⊗IIZ + IIX⊗D2y⊗IIZ + IIX⊗IIY⊗D2z
    return ∇2
end
```

The main difference is that we have used the symbol  $\otimes$  in place of kron function to make our code simpler. In Julia this symbol can be entered by typing `\otimes` in Julia console or text editors that have been set up to use Julia extension and Unicode input.

## 4.1 3d harmonic oscillator

We hope that at this point you will have no difficulties to create your own Schrodinger equation solver for 3d harmonic potential:

$$V(x, y, z) = \frac{1}{2}\omega^2 (x^2 + y^2 + z^2) \quad (4.4)$$

The analytic solution for the energy levels of this potential can be written as:

$$E_{n_x+n_y+n_z} = \hbar\omega \left( n_x + n_y + n_z + \frac{3}{2} \right) \quad (4.5)$$

where  $n_x, n_y, n_z$  are integers positive integers including zero. For each energy level we have the following degeneracies:

$$g_n = \frac{(n+1)(n+2)}{2} \quad (4.6)$$

where  $n = n_x + n_y + n_z$ . For examples:

```
n = n_x + n_y + n_z

n = 0: (1)(2)/2 = 1   (E=1.5)
n = 1: (2)(3)/2 = 3   (E=2.5)
n = 2: (3)(4)/2 = 6   (E=3.5)
```

The following is the result of `diag_Emin_PCG`, for the 4 lowest eigenstates:

```
CG step      12 =      8.9999996162 4.4341014e-06
evals[ 1] =      1.4999999732, devals =      4.1020002284e-07
evals[ 2] =      2.4999998628, devals =      1.0054977597e-06
evals[ 3] =      2.4999998760, devals =      1.3278320172e-06
evals[ 4] =      2.4999999042, devals =      1.6905715712e-06
iter 12 nconv = 4
Convergence is achieved based on nconv
```

The obtained eigenvalues are close the the analytic solution. Note that the we have second lowest eigenstate of degeneracy 3. You can try to vary the number of requested eigenstates to very that the degeneracies are correct.

## 4.2 Hydrogen atom

### 4.2.1 Coulomb potential and its singularity

Until now, we only have considered simple potentials such as harmonic potential. Now we will move on and consider more realistic potentials which is used in practical electronic calculations.

For most applications in materials physics and chemistry the external potential that is felt by electrons is the Coulombic potential due to atomic nucleus. This potential has the following form:

$$V(r) = - \sum_I^{N_{\text{atom}}} \frac{Z_I}{|\mathbf{r} - \mathbf{R}_I|} \quad (4.7)$$

where  $R_l$  are the positions and  $Z_l$  are the charges of the atomic nucleus present in the system. We will consider the most simplest system, namely the hydrogen atom  $Z_l = 1$ , for which we have

$$V(r) = -\frac{1}{|\mathbf{r} - \mathbf{R}_0|} \quad (4.8)$$

The following Julia code implement the H atom potential:

```
function pot_H_atom( grid; r0=(0.0, 0.0, 0.0) )
    Npoints = grid.Npoints
    Vpot = zeros(Npoints)
    for i in 1:Npoints
        dx = grid.r[1,i] - r0[1]
        dy = grid.r[2,i] - r0[2]
        dz = grid.r[3,i] - r0[3]
        Vpot[i] = -1.0/sqrt(dx^2 + dy^2 + dz^2)
    end
    return Vpot
end
```

With only minor modification to our program for harmonic potential, we can solve the Schroedinger equation for the hydrogen atom:

```
grid = FD3dGrid( (-5.0,5.0), Nx, (-5.0,5.0), Ny, (-5.0,5.0), Nz )
∇2 = build_nabla2_matrix( grid )
Vpot = pot_H_atom( grid )
Ham = -0.5*∇2 + spdiag( 0 => Vpot )
prec = aspreconditioner(ruge_stuben(Ham))
Nstates = 1 # only choose the lowest lying state
Npoints = Nx*Ny*Nz
X = ortho_sqrt( rand(Float64, Npoints, Nstates) ) # random initial guess of wave function
evals = diag_LOBPCG!( Ham, X, prec, verbose=true )
```

For the grid size of  $N_x = N_y = N_z = 50$  and using 9-point finite-difference approximation to the second derivative operator in 1d we obtain the eigenvalue of -0.4900670759 Ha which is not too bad if compared with the exact value of -0.5 Ha. We can try to increase the grid size until we can get satisfactory result.

Note that there is a caveat when we are trying to use the Coulombic potential. This potential is divergent at  $r = 0$ , so care must be taken such that this divergence is not encountered in the numerical calculation. In the previous code, we have tried to achieve this by using choosing the numbers  $N_x$ ,  $N_y$ , and  $N_z$  to be even numbers. This way, we avoid the evaluation of the potential at the singular point of the Coulomb potential (the point  $\mathbf{r} = (0, 0, 0)$  in this particular case).

### 4.2.2 Pseudopotential

In the many electronic structure calculations, it is sometime convenient to replace the Coulomb potential with another potential which is smoother which we will refer to as a **pseudopotential**. Not all smooth potentials can do the job. The smooth potential should satisfy several requirements. One of the most important requirement is that the smooth potential should have similar scattering properties as the original Coulomb potential that it replaces. This means that the potential should have posses similar eigenvalues as the original Coulomb potential. Usually we don't try to reproduce all eigenvalue spectrum but only the eigenvalues which belongs to the valence electrons. The valence electrons are responsible for most chemically and physically important properties so this is an acceptable approximation for most cases.

The theory and algorithms for constructing pseudopotentials are beyond the scope of this book. Most pseudopotentials are non-local by construction and this can make our program rather complicated. In this chapter we will focus on the so-called local pseudopotential. Practically, local pseudopotentials pose no additional difficulties as the potentials that we have considered so far. As an example of a pseudopotential, we will consider the following local pseudopotential for hydrogen atom:

$$V_{H,ps}(r) = -\frac{Z_{val}}{r} \operatorname{erf}\left(\frac{\bar{r}}{\sqrt{2}}\right) + \exp\left(-\frac{1}{2}\bar{r}^2\right) (C_1 + C_2\bar{r}^2) \quad (4.9)$$

where  $\bar{r} = r/r_{loc}$  and with the parameters  $r_{loc} = 0.2$ ,  $Z_{val} = 1$ ,  $C_1 = -4.0663326$ , and  $C_2 = 0.6678322$ .

The following Julia code implements this pseudopotential.

```
function pot_Hps_HGH( grid; r0=(0.0, 0.0, 0.0) )
    Npoints = grid.Npoints
    Vpot = zeros( Float64, Npoints )
    Zval = 1
    rloc = 0.2
    C1 = -4.0663326
    C2 = 0.6678322
    for ip = 1:Npoints
        dx2 = ( grid.r[1,ip] - r0[1] )^2
        dy2 = ( grid.r[2,ip] - r0[2] )^2
        dz2 = ( grid.r[3,ip] - r0[3] )^2
        r = sqrt(dx2 + dy2 + dz2)
        if r < eps()
            Vpot[ip] = -2*Zval/(sqrt(2*pi)*rloc) + C1
        else
            rrloc = r/rloc
            Vpot[ip] = -Zval/r * erf( r/(sqrt(2.0)*rloc) ) + (C1 +
                ↪ C2*rrloc^2)*exp(-0.5*(rrloc)^2)
        end
    end
    return Vpot
end
```

Note that, you need to import erf from the package SpecialFunctions:

```
import SpecialFunctions: erf
```

We also have been careful to not evaluate the term  $Z_{val}/r$  by adding a check for the value of  $r$ . If it is very small (close to zero, or smaller than  $\operatorname{eps}()$ ) then we are using the limiting value of 4.9 when  $r \rightarrow 0$ :

$$V_{H,ps}(r) = -\frac{Z_{val}}{r} \operatorname{erf}\left(\frac{r}{\sqrt{2}r_{loc}}\right) + C_1 \quad (4.10)$$

$$V_{H,ps}(\sqrt{2}r_{loc}) = -\frac{Z_{val}}{r\sqrt{2}r_{loc}} \frac{\operatorname{erf}(r)}{r} + C_1 \quad (4.11)$$

$$V_{H,ps}(0) = -\frac{Z_{val}}{\sqrt{2}r_{loc}} \frac{2}{\sqrt{\pi}} + C_1 \quad (4.12)$$

The comparison of this potential and Coulomb potential can be seen in Figure 4.1.






Figure 4.1: Comparison Coulomb potential vs pseudopotential

The pseudopotential in 4.9 is actually a special case of the more general form of the local part of pseudopotential developed by Goedecker-Teter-Hutter (GTH):

$$V_{\text{loc}}^{\text{ps}}(r) = -\frac{Z_{\text{val}}}{r} \text{erf}\left(\frac{\bar{r}}{\sqrt{2}}\right) + \exp\left(-\frac{1}{2}\bar{r}^2\right) (C_1 + C_2\bar{r}^2 + C_3\bar{r}^4 + C_4\bar{r}^6) \quad (4.13)$$

GTH pseudopotentials also come with the nonlocal potential which is considerably more complicated than the local one. We will consider the case of nonlocal pseudopotentials in Chapter 7. GTH pseudopotentials have been parameterized from density functional calculations for many elements in the periodic table.

### 4.3 Exercises

Compare the convergence of eigenvalue of hydrogen potential when using Coulomb potential vs pseudopotential.

Higher eigenstates of H atom + visualization using Xcrysden

H<sub>2</sub> molecule

He atom

LiH molecule



# Chapter 5

## Poisson equation

### 5.1 Conjugate gradient method

In this section we will discuss a second type of equation that is important in solving Kohn-Sham equation, namely the Poisson equation. In the context of solving Kohn-Sham equation, Poisson equation is used to calculate classical electrostatic potential due to some electronic charge density. The Poisson equation that we will solve have the following form:

$$\nabla^2 V_{\text{Ha}}(\mathbf{r}) = -4\pi\rho(\mathbf{r}) \quad (5.1)$$

where  $\rho(\mathbf{r})$  is the electronic density. Using finite difference discretization for the operator  $\nabla^2$  we end up with the following linear equation:

$$\mathbb{L}\mathbf{V} = \mathbf{f} \quad (5.2)$$

where  $\mathbb{L}$  is the matrix representation of the Laplacian operator  $\mathbf{f}$  is the discrete representation of the right hand side of the equation 5.1, and the unknown  $\mathbf{V}$  is the discrete representation of the Hartree potential.

There exist several methods for solving the linear equation 5.2. We will use the so-called conjugate gradient method for solving this equation. This method is an iterative method, so it generally needs a good preconditioner to achieve good convergence. A detailed derivation about the algorithm is beyond this article and the readers are referred to several existing literatures [9, 10] and a webpage [11] for more information. The algorithm is described in `Poisson_solve_PCG.jl`

```
function Poisson_solve_PCG( Lmat, prec, f; NiterMax=1000 TOL=5.e-10 )
    Npoints = size(f,1)
    phi = zeros( Float64, Npoints )
    r = zeros( Float64, Npoints )
    p = zeros( Float64, Npoints )
    z = zeros( Float64, Npoints )
    nabla2_phi = Lmat*phi
    r = f - nabla2_phi
    z = copy(r)
    ldiv!(prec, z)
    p = copy(z)
    rsold = dot( r, z )
    for iter = 1 : NiterMax
        nabla2_phi = Lmat*p
        alpha = rsold/dot( p, nabla2_phi )
        phi = phi + alpha * p
        r = r - alpha * nabla2_phi
    end
end
```

```

z = copy(r)
ldiv!(prec, z)
rsnew = dot(z, r)
deltars = rsold - rsnew
if sqrt(abs(rsnew)) < TOL
    break
end
p = z + (rsnew/rsold) * p
rsold = rsnew
end
return phi
end

```

To test our implementation we will adopt a problem given in Prof. Arias Practical DFT mini-course [12]. In this problem we will solve Poisson equation for a given charge density built from superposition of two Gaussian charge density:

$$\rho(\mathbf{r}) = \frac{1}{(2\pi\sigma_1^2)^{\frac{3}{2}}} \exp\left(-\frac{\mathbf{r}^2}{2\sigma_1^2}\right) - \frac{1}{(2\pi\sigma_2^2)^{\frac{3}{2}}} \exp\left(-\frac{\mathbf{r}^2}{2\sigma_2^2}\right) \quad (5.3)$$

After we obtain  $V_{\text{Ha}}(\mathbf{r})$ , we calculate the Hartree energy:

$$E_{\text{Ha}} = \frac{1}{2} \int \rho(\mathbf{r}) V_{\text{Ha}}(\mathbf{r}) d\mathbf{r} \quad (5.4)$$

and compare the result with the analytical formula.

```

function test_main( NN::Array{Int64} )
    AA = [0.0, 0.0, 0.0]
    BB = [16.0, 16.0, 16.0]
    # Initialize grid
    FD = FD3dGrid( NN, AA, BB )
    # Box dimensions
    Lx = BB[1] - AA[1]
    Ly = BB[2] - AA[2]
    Lz = BB[3] - AA[3]
    # Center of the box
    x0 = Lx/2.0
    y0 = Ly/2.0
    z0 = Lz/2.0
    # Parameters for two gaussian functions
    sigma1 = 0.75
    sigma2 = 0.50

    Npoints = FD.Nx * FD.Ny * FD.Nz
    rho = zeros(Float64, Npoints)
    phi = zeros(Float64, Npoints)
    # Initialization of charge density
    dr = zeros(Float64, 3)
    for ip in 1:Npoints
        dr[1] = FD.r[1,ip] - x0
        dr[2] = FD.r[2,ip] - y0
        dr[3] = FD.r[3,ip] - z0
        r = norm(dr)
        rho[ip] = exp( -r^2 / (2.0*sigma2^2) ) / (2.0*pi*sigma2^2)^1.5 -
            exp( -r^2 / (2.0*sigma1^2) ) / (2.0*pi*sigma1^2)^1.5
    end
end

```

```

end
deltaV = FD.hx * FD.hy * FD.hz
Laplacian3d = build_nabla2_matrix( FD, func_ld=build_D2_matrix_9pt )
prec = aspreconditioner(ruge_stuben(Laplacian3d))
@printf("Test norm charge: %18.10f\n", sum(rho)*deltaV)
print("Solving Poisson equation:\n")
phi = Poisson_solve_PCG( Laplacian3d, prec, -4*pi*rho, 1000, verbose=true, TOL=1e-10 )
# Calculation of Hartree energy
Unum = 0.5*sum( rho .* phi ) * deltaV
Uana = ((1.0/sigma1 + 1.0/sigma2 )/2.0 - sqrt(2.0)/sqrt(sigma1^2 + sigma2^2))/sqrt(pi)
@printf("Numeric = %18.10f\n", Unum)
@printf("Uana = %18.10f\n", Uana)
@printf("abs diff = %18.10e\n", abs(Unum-Uana))
end
test_main([64,64,64])

```

Result:

```

Numeric =      0.0551434259
Uana     =      0.0551425277
abs diff =  8.9818466372e-07

```

FIXME: Needs correction, boundary condition is not treated properly.

## 5.2 Reciprocal space method

Using FFT, natural choice for periodic boundary condition

## 5.3 Direct integration

Ref: D. Sundholm. Universal method for computation of electrostatic potentials J. Chem. Phys. 122 194107 (2005)

S.A. Losilla, D. Sundholm, and J. Juselius. Direct approach to gravitation and electrostatics method for periodic systems J. Chem. Phys. 132 024102 (2010)

$$V(\mathbf{r}_1) = \int_{-\infty}^{+\infty} \rho(\mathbf{r}_2) \frac{1}{r_{12}} d\mathbf{r}_2 \quad (5.5)$$

Rewriting Coulomb operator as integral expression using:

$$\frac{1}{r} = \frac{2}{\sqrt{\pi}} \int_0^{\infty} e^{-r^2 t^2} dt \quad (5.6)$$

$$V(\mathbf{r}_1) = \frac{2}{\sqrt{\pi}} \int_0^{\infty} \int_{-\infty}^{+\infty} e^{-t^2(r_1-r_2)^2} \rho(\mathbf{r}_2) d\mathbf{r}_2 dt \quad (5.7)$$

Numerical tensorial basis:

$$\rho(\mathbf{r}_2) = \sum_{\alpha\beta\gamma} d_{\alpha\beta\gamma} \chi_{\alpha}(x_2) \chi_{\beta}(y_2) \chi_{\gamma}(z_2) \quad (5.8)$$

$$\begin{aligned}
V_0(x_1, y_1, z_1) = \frac{2}{\sqrt{p!}} \sum_{\alpha_t} w_{\alpha_t} \sum_{\alpha\beta\gamma} d_{\alpha\beta\gamma} \int_{-\infty}^{\infty} e^{-t_{\alpha_t}^2(z_1-z_2)} \chi_{\gamma}(z_2) dz_2 \times \int_{-\infty}^{\infty} e^{-t_{\alpha_t}^2(y_1-y_2)} \chi_{\beta}(y_2) dy_2 \\
\times \int_{-\infty}^{\infty} e^{-t_{\alpha_t}^2(x_1-x_2)} \chi_{\alpha}(x_2) dx_2 \quad (5.9)
\end{aligned}$$

$t_{\alpha_t}$ : integration points,  $w_{\alpha_t}$ : integration weights,

$$F_{\gamma_x \alpha_t}^{x, \alpha_x} = \int_{-\infty}^{\infty} e^{-t_{\alpha_t}^2(x_{\alpha_x}-x_2)^2} \chi_{\gamma_x}(x_2) dx_2 \quad (5.10)$$

$$V_{\alpha_x \alpha_y \alpha_z} = \frac{2}{\sqrt{\pi}} \sum_{\alpha_t} w_{\alpha_t} \sum_{\gamma_z} F_{\gamma_z \alpha_z}^{z, \alpha_t} \sum_{\gamma_y} F_{\gamma_y \alpha_y}^{y, \alpha_t} \sum_{\gamma_x} F_{\gamma_x \alpha_x}^{x, \alpha_t} d_{\gamma_x \gamma_y \gamma_z} \quad (5.11)$$

## Chapter 6

# Kohn-Sham equation part I

In this chapter we will put together tools that we have built in the previous chapters. Our task is to solve the Kohn-Sham equation:

$$\left[ -\frac{1}{2}\nabla^2 + V_{\text{KS}}(\mathbf{r}) \right] \psi_i(\mathbf{r}) = \epsilon_i \psi_i(\mathbf{r}) \quad (6.1)$$

where  $V_{\text{KS}}(\mathbf{r})$  is effective single particle potential or the Kohn-Sham potential:

$$V_{\text{KS}}(\mathbf{r}) = V_{\text{ext}}(\mathbf{r}) + V_{\text{Ha}}(\mathbf{r}) + V_{\text{xc}}(\mathbf{r}) \quad (6.2)$$

The Kohn-Sham equation looks very much like Schroedinger equation, but with two additional potentials: the Hartree and XC potential. We will consider a 3d systems, so the techniques that we have learned in Chapter 4 will be used for solving the Kohn-Sham equation. Due to the presence of Hartree and XC potential, there will be additional steps that need to be performed in order to solve the Kohn-Sham equation. To calculate Hartree and XC potential, we need to calculate electron density which depend on the solution of the Kohn-Sham equation itself. Because of this chicken-and-egg characteristic, the Kohn-Sham equation must be solved self-consistently.

Solutions to the Kohn-Sham equation are the Kohn-Sham orbitals  $\psi_i(\mathbf{r})$  and eigenvalues  $\epsilon_i$ . Total Kohn-Sham energy can be written as:

$$E_{\text{tot}}^{\text{KS}} = E_{\text{kin}} + E_{\text{ext}} + E_{\text{Ha}} + E_{\text{xc}} \quad (6.3)$$

The kinetic energy of noninteracting electrons:

$$E_{\text{kin}} = -\frac{1}{2} \sum_i \int \psi_i^*(\mathbf{r}) \nabla^2 \psi_i(\mathbf{r}) d\mathbf{r} \quad (6.4)$$

External potential energy:

$$E_{\text{ext}} = \int \rho(\mathbf{r}) V_{\text{ext}}(\mathbf{r}) d\mathbf{r} \quad (6.5)$$

Hartree energy:

$$E_{\text{ext}} = \int \rho(\mathbf{r}) V_{\text{Ha}}(\mathbf{r}) d\mathbf{r} \quad (6.6)$$

XC energy (LDA):

$$E_{\text{ext}} = \int \epsilon_{\text{xc}}[\rho(\mathbf{r})] \rho(\mathbf{r}) d\mathbf{r} \quad (6.7)$$

where  $\epsilon_{\text{xc}}$  is the XC energy per particle per volume.

Alternative expression of total energy as sum of orbital energies:

$$E_{\text{tot}}^{\text{KS}} = \sum_i^N \epsilon_i - E_{\text{Ha}} + E_{\text{xc}} - \int \frac{\delta E_{\text{xc}}}{\delta \rho(\mathbf{r})} \rho(\mathbf{r}) d\mathbf{r} \quad (6.8)$$

Nucleus-nucleus interaction energy:

$$E_{\text{tot}} = E_{\text{tot}}^{\text{KS}} + E_{\text{NN}} \quad (6.9)$$

We first consider that the case of  $V_{\text{xc}} = 0$ . This will make our first implementation of self-consistent field to be easier.

## 6.1 Hartree calculation

Hartree potential is the classical electrostatic interaction potential. It is defined as

$$V_{\text{Ha}}(\mathbf{r}) = \int \frac{\rho(\mathbf{r}')}{|\mathbf{r} - \mathbf{r}'|} d\mathbf{r}' \quad (6.10)$$

Given the electron density:

$$\rho(\mathbf{r}) = \sum_i f_i \psi_i^*(\mathbf{r}) \psi_i(\mathbf{r}) \quad (6.11)$$

we can use any methods in Chapter 5 to calculate  $V_{\text{Ha}}(\mathbf{r})$ . The factor  $f_i$  in the Equation 6.11 is the occupation number of the electron.

### 6.1.1 The Hamiltonian struct

Because our Hamiltonian contain additional potential, its action to a wave function is more complicated. We also need to implement several operations such as calculation of electron density for given wave functions and total energy. These calculations usually need access to several variables. To make access to various variables easy, we will wrap them in a "big" Julia struct. We will call this struct as Hamiltonian.

The definition of the Hamiltonian struct can be seen in the following Julia code.

```
mutable struct Hamiltonian
  grid::Union{FD3dGrid,LF3dGrid}
  Laplacian::SparseMatrixCSC{Float64,Int64}
  V_Ps_loc::Vector{Float64}
  V_Hartree::Vector{Float64}
  electrons::Electrons
  rhoe::Vector{Float64}
  atoms::Atoms
  precKin
  psolver::Union{PoissonSolverDAGE,PoissonSolverFFT}
  energies::Energies
  gvec::Union{Nothing,GVectors}
end
```

A short explanation about the fields of Hamiltonian follows.

- `grid`: the field that describes the real space grid points or basis functions that is used to represent quantities such as wave functions, potentials, and densities. In our case, this field is an instance of `FD3dGrid` or `LF3dGrid`.



- Laplacian: the matrix representation of the Laplacian operator
- V\_Ps\_loc: the local pseudopotential. It also will represent the any external local potential that is felt by electrons. Note that we have chosen the name V\_Ps\_loc because of the generalization that we will do in Chapter 7
- V\_Hartree: the Hartree potential.
- electrons: an instance of type Electrons. This field stores various variables that are used to describe electron states such as number of electrons, number of states, occupation numbers, etc.
- rhoe: the electron density
- atoms: an instance of Atoms. This field can be used to represent molecules, for example.
- precKin: the preconditioner based on the kinetic matrix. This preconditioner can be for diagonalization of energy minimization.
- psolver: the Poisson equation solver which is used to calculate the Hartree potential.
- energies: an instance of Energies. This field stores various total energy components.
- gvec: an instance of GVectors. This field is only relevant for periodic structure.

The following constructor can be used to initialize an instance of Hamiltonian.

```
function Hamiltonian(
    atoms::Atoms, grid, V_Ps_loc;
    Nelectrons=2, Nstates_extra=0
)
    Laplacian = build_nabla2_matrix( grid )
    if grid.pbc == (true,true,true)
        gvec = GVectors(grid)
    else
        gvec = nothing
    end
    Npoints = grid.Npoints
    V_Hartree = zeros(Float64, Npoints)
    Rhoe = zeros(Float64, Npoints)
    precKin = aspreconditioner( ruge_stuben(-0.5*Laplacian) )
    electrons = Electrons( Nelectrons, Nstates_extra=Nstates_extra )
    if grid.pbc == (false,false,false)
        psolver = PoissonSolverDAGE(grid)
    else
        psolver = PoissonSolverFFT(grid)
    end
    energies = Energies()
    return Hamiltonian( grid, Laplacian, V_Ps_loc, V_Hartree, electrons,
                       Rhoe, atoms, precKin, psolver, energies, gvec )
end
```

An important operation that must be defined for Hamiltonian is the multiplication between Hamiltonian and wave function. We can define this operator by overloading the `*` operator. In the following code, first we apply the kinetic matrix by using the Laplacian field of Hamiltonian, using sparse matrix multiplication. This step is the followed by application of potential operator which now consists of V\_Ps\_loc and V\_Hartree.

```

import Base: *
function *( Ham::Hamiltonian, psi::Matrix{Float64} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    Hpsi = zeros(Float64,Nbasis,Nstates)
    Hpsi = -0.5 * Ham.Laplacian * psi
    for ist in 1:Nstates, ip in 1:Nbasis
        Hpsi[ip,ist] = Hpsi[ip,ist] + ( Ham.V_Ps_loc[ip] + Ham.V_Hartree[ip] ) * psi[ip,ist]
    end
    return Hpsi
end

```

We also define a function for updating the potential for an input electron density. For our current purpose, this function do two things: copy the input electron density to Ham.rhoe and calculate the Hartree potential by calling Poisson\_solve function.

```

function update!( Ham::Hamiltonian, Rhoe::Vector{Float64} )
    Ham.rhoe = Rhoe
    Ham.V_Hartree = Poisson_solve( Ham.psolver, Ham.grid, Rhoe )
    return
end

```

### 6.1.2 Electrons

The type Electrons stores several variables related to single-electron states.

```

mutable struct Electrons
    Nelectrons::Int64
    Nstates::Int64
    Nstates_occ::Int64
    Focc::Array{Float64,1}
    ene::Array{Float64,1}
end

```

Electron density calculation 6.11:

```

function calc_rhoe( Ham, psi::Array{Float64,2} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    Rhoe = zeros(Float64,Nbasis)
    for ist in 1:Nstates
        f = Ham.electrons.Focc[ist]
        for ip in 1:Nbasis
            Rhoe[ip] = Rhoe[ip] + f*psi[ip,ist]*psi[ip,ist]
        end
    end
    return Rhoe
end

```

### 6.1.3 Total energy terms

We introduce the Energies type to store various energy terms.

```
mutable struct Energies
    Kinetic::Float64
    Ps_loc::Float64
    Hartree::Float64
    NN::Float64
end
```

Overload sum:

```
import Base: sum
function sum( ene::Energies )
    return ene.Kinetic + ene.Ps_loc + ene.Hartree + ene.NN
end
```

Calculation of the kinetic energy:

$$E_{\text{kin}} = -\frac{1}{2} \int \psi_i(\mathbf{r}) \nabla^2 \psi_i(\mathbf{r}) d\mathbf{r} \quad (6.12)$$

can be done using the following function.

```
function calc_E_kin( Ham::Hamiltonian, psi::Array{Float64,2} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    E_kin = 0.0
    nabla2psi = zeros(Float64,Nbasis)
    dVol = Ham.grid.dVol
    for ist in 1:Nstates
        @views nabla2psi = -0.5*Ham.Laplacian*psi[:,ist]
        @views E_kin = E_kin + Ham.electrons.Focc[ist]*dot( psi[:,ist], nabla2psi[:] )*dVol
    end
    return E_kin
end
```

In the function `calc_energies!`, we calculate total electronic energy terms. This function modifies `Ham.energies`.

```
function calc_energies!( Ham::Hamiltonian, psi::Array{Float64,2} )
    dVol = Ham.grid.dVol
    Ham.energies.Kinetic = calc_E_kin( Ham, psi )
    Ham.energies.Ps_loc = sum( Ham.V_Ps_loc .* Ham.rhoe )*dVol
    Ham.energies.Hartree = 0.5*sum( Ham.V_Hartree .* Ham.rhoe )*dVol
    return
end
```

It is sometimes convenient to return the energies directly. This is done by the function `calc_energies` which just wraps `calc_energies!` function.

```
function calc_energies(Ham, psi)
    calc_energies!(Ham, psi)
    return Ham.energies
end
```

### 6.1.4 Harmonic potential

We are now ready to implement our self-consistent field solution to the Kohn-Sham equation. As in the previous chapters, we will chose a system with 3d harmonic potential as the external potential.

First, we define the grid.

```
AA = [-3.0, -3.0, -3.0]
BB = [3.0, 3.0, 3.0]
NN = [25, 25, 25]
grid = FD3dGrid( NN, AA, BB )
```

The, we calculate the external potential. We will use `V_Ps_loc` as the name of the external potential. We also choose  $\omega = 2$ .

```
V_Ps_loc = pot_harmonic( grid,  $\omega=2$  )
```

The next step is to initialize an instance of Hamiltonian. We need to specify number of electrons and number of states for our electronic states. Here we choose to 8 electrons and 4 states, each states is doubly occupied.

```
Nelectrons = 8
Nstates = round(Int64,Nelectrons/2)
Ham = Hamiltonian( Atoms(), grid, V_Ps_loc, Nelectrons=Nelectrons )
```

We prepare random wave functions for guess solution to the Kohn-Sham equation.

```
Npoints = grid.Npoints
dVol = grid.dVol
psi = rand(Float64,Npoints,Nstates)
ortho_sqrt!(psi)
psi = psi/sqrt(dVol)
```

and calculate the electron density associated with these wave functions. We also printed the integrated electron density, approximated with simple summation. Note that the integrated electron density should be close the the number of electrons.

```
Rhoe = calc_rhoe( Ham, psi )
@printf("Integrated Rhoe = %18.10f\n", sum(Rhoe)*dVol)
```

From the guess electron density, we update the Hamiltonian, calculate the Hartree potential by calling the `update!` function.

```
update!( Ham, Rhoe )
```

We also calculate total energy for our initial guess wave functions and electron density.

```
Etot = sum( calc_energies( Ham, psi ) )
@printf("Initial Etot = %18.10f\n", Etot)
```

Before entering the SCF cycle, we need to prepare and define several variables which will be used in the SCF cycle.

```

evals = zeros(Float64,Nstates)
Etot_old = Etot
dEtot = 0.0
dRhoe = 0.0
betamix = 0.5
NiterMax = 100

```

An important variable that needs attention is `betamix`. This variable plays the role of  $\beta$  in the following equation.

$$\rho_{\text{in}}^{i+1}(\mathbf{r}) = \beta \rho(\mathbf{r})_{\text{out}}^i + (1 - \beta) \rho_{\text{in}}^i(\mathbf{r}) \quad (6.13)$$

where  $0 < \beta \leq 1$ . In the Equation (6.13),  $\rho_{\text{in}}^{i+1}(\mathbf{r})$  is the input density for the next SCF iteration.

The SCF cycle is implemented in the following Julia code.

```

for iterSCF in 1:NiterMax
    evals = diag_LOBPCG!(Ham, psi, Ham.precKin)
    psi = psi/sqrt(dVol)
    Rhoe_new = calc_rhoe(Ham, psi)
    Rhoe = betamix*Rhoe_new + (1-betamix)*Rhoe
    update!( Ham, Rhoe )
    Etot = sum( calc_energies( Ham, psi ) )
    dRhoe = sum(abs.(Rhoe - Rhoe_new))/Npoints
    dEtot = abs(Etot - Etot_old)
    @printf("%5d %18.10f %18.10e %18.10e\n", iterSCF, Etot, dEtot, dRhoe)
    if dEtot < 1e-6
        @printf("Convergence is achieved in %d iterations\n", iterSCF)
        for i in 1:Nstates
            @printf("%3d %18.10f\n", i, evals[i])
        end
        break
    end
    Etot_old = Etot
end

```

The result of the SCF:

```

... snipped
 33    62.2907110243    1.4097586885e-06    5.3155705598e-08
 34    62.2907100832    9.4107285520e-07    4.5940110326e-08
Convergence is achieved in 34 iterations
 1      9.8038626839
 2     11.1505190947
 3     11.1505191402
 4     11.1505192492
-----
Total energy components
-----
Kinetic =      12.9807771382
Ps_loc  =      25.0898019726
Hartree =      24.2201309724
NN       =           0.0000000000
-----
Total    =      62.2907100832

```

### 6.1.5 H atom

A next example that we will try is hydrogen atom. Here is our setup.

```
AA = [-8.0, -8.0, -8.0]
BB = [ 8.0,  8.0,  8.0]
NN = [41, 41, 41]
grid = FD3dGrid( NN, AA, BB )
atoms = Atoms( xyz_string=
    """
    1

    H  0.0  0.0  0.0
    """ )
V_Ps_loc = pot_Hps_HGH(atoms, grid)
Nstates = 1
Nelectrons = 1
Ham = Hamiltonian(atoms, grid, V_Ps_loc, Nelectrons=1)
```

Result:

```
... snipped
14      -0.2562610358    2.0600214368e-06    1.9590484379e-08
15      -0.2562602740    7.6184773828e-07    9.7155909123e-09
Convergence is achieved in 15 iterations
1      -0.0428298539
-----
Total energy components
-----
Kinetic =      0.2928408518
Ps_loc  =     -0.7625329979
Hartree =      0.2134318721
NN       =      0.0000000000
-----
Total   =     -0.2562602740
```

TODO: try using denser grid. Study convergence of the total energy with respect grid spacing.

### 6.1.6 H<sub>2</sub> molecule

Using the same grid as we have used for hydrogen atom.

```
atoms = Atoms( xyz_string=
    """
    2

    H  0.75  0.0  0.0
    H -0.75  0.0  0.0
    """, in_bohr=true)
V_Ps_loc = pot_Hps_HGH(atoms, grid)
Nstates = 1
Nelectrons = 2
Ham = Hamiltonian(atoms, grid, V_Ps_loc, Nelectrons=Nelectrons)
```

We need to calculate ion-ion interaction energy. Do this before SCF cycle.

```
Ham.energies.NN = calc_E_NN( [1.0, 1.0], atoms.positions )
```

Result:

```
... snipped
16      -0.5972202712    1.3097904592e-06    1.0969070600e-08
17      -0.5972197417    5.2951699026e-07    5.4868596190e-09
Convergence is achieved in 17 iterations
1       -0.1083633271
-----
Total energy components
-----
Kinetic =      0.8229371585
Ps_loc  =     -3.1339854624
Hartree =      1.0471618956
NN       =      0.6666666667
-----
Total    =     -0.5972197417
```

TODO: try using denser grid. Study convergence of the total energy with respect grid spacing.

## 6.2 Kohn-Sham calculations

We are now ready to include the XC term.

$$E_{xc}[\rho(\mathbf{r})] = \int \rho(\mathbf{r}) \varepsilon_{xc}(\rho(\mathbf{r})) d\mathbf{r} \quad (6.14)$$

Decomposition:

$$E_{xc} = E_x + E_c \quad (6.15)$$

### 6.2.1 Exchange term: Slater exchange

Homogeneous electron, exchange energy:

$$E_x[\rho(\mathbf{r})] = -\frac{3}{4} \left( \frac{3}{\pi} \right)^{1/3} \int \rho(\mathbf{r})^{4/3} d\mathbf{r} \quad (6.16)$$

Or can be written as:

$$E_x[\rho(\mathbf{r})] = \int \rho(\mathbf{r}) \varepsilon_x(\rho(\mathbf{r})) d\mathbf{r} \quad (6.17)$$

Exchange energy density:

$$\varepsilon_x(\rho) = C_x \rho^{1/3} \quad (6.18)$$

with parameter:

$$C_x = \frac{3}{4} \left( \frac{3}{\pi} \right)^{1/3} \quad (6.19)$$

### 6.2.2 Correlation VWN

VWN parameterization:

$$\varepsilon_c = A \left\{ \ln \frac{x}{X(x)} + \frac{2b}{Q} \operatorname{atan} \frac{Q}{2x+b} - \frac{bx_0}{X(x_0)} \left[ \ln \frac{(x-x_0)^2}{X(x)} + \frac{2(b+2x_0)}{Q} \operatorname{atan} \frac{Q}{2x+b} \right] \right\} \quad (6.20)$$

$$x = \sqrt{r_s} \quad (6.21)$$

$$r_s = \left( \frac{3}{4\pi\rho} \right)^{1/3} \quad (6.22)$$

$$X(x) = x^2 + bx + c \quad (6.23)$$

$$Q = \sqrt{4c - b^2} \quad (6.24)$$

Parameter:

$$A = 0.0310907 \quad (6.25)$$

$$b = 3.72744 \quad (6.26)$$

$$c = 12.9352 \quad (6.27)$$

$$x_0 = -0.10498 \quad (6.28)$$

Wigner-Seitz radius is defined by:

$$\frac{4}{3}\pi r_s^3 = \frac{1}{\rho} \quad (6.29)$$

Potential:

$$V_{xc}^{LDA} = \frac{\delta E_{xc}^{LDA}}{\delta \rho(\mathbf{r})} = \epsilon_{xc}(\rho(\mathbf{r})) + \rho(\mathbf{r}) \frac{\partial \epsilon_{xc}(\rho(\mathbf{r}))}{\partial \rho(\mathbf{r})} \quad (6.30)$$

### 6.2.3 Implementation

New Hamiltonian, include  $V_{XC}$ .

Update the potential:

```
function update!( Ham::Hamiltonian, Rhoe::Vector{Float64} )
    Ham.rhoe = Rhoe
    Ham.V_Hartree = Poisson_solve( Ham.psolver, Ham.grid, Rhoe )
    Ham.V_XC = calc_Vxc_VWN( XC_Calculator(), Rhoe )
    return
end
```

Application of Hamiltonian

```
import Base: *
function *( Ham::Hamiltonian, psi::Matrix{Float64} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    Hpsi = zeros{Float64,Nbasis,Nstates}
    Hpsi = -0.5*Ham.Laplacian * psi
    for ist in 1:Nstates, ip in 1:Nbasis
        Hpsi[ip,ist] = Hpsi[ip,ist] + ( Ham.V_Ps_loc[ip] + Ham.V_Hartree[ip] +
            Ham.V_XC[ip] ) * psi[ip,ist]
    end
    return Hpsi
end
```

Calculation of XC energy:



```
psxc = calc_epsxc_VWN( XCcalculator(), Ham.rhoe )
Ham.energies.XC = sum( epsxc .* Ham.rhoe )*dVol
```

### 6.2.4 Harmonic potential

Setup:

```
AA = [-3.0, -3.0, -3.0]
BB = [3.0, 3.0, 3.0]
NN = [25, 25, 25]
grid = FD3dGrid( NN, AA, BB )
V_Ps_loc = pot_harmonic( grid, w=2 )
Nelectrons = 8
Nstates = round(Int64,Nelectrons/2)
Ham = Hamiltonian( Atoms(), grid, V_Ps_loc, Nelectrons=Nelectrons )
```

```
... snipped
 19      57.5543743808      1.0560459387e-06      4.0259990163e-05
 20      57.5543736562      7.2459251044e-07      3.1156747100e-05
Convergence is achieved in 20 iterations
```

Eigenvalues:

```
1      9.0567277414
2     10.4655966931
3     10.4655970965
4     10.4655977729
```

-----  
Total energy components

```
-----
Kinetic =      13.6600637584
Ps_loc  =      23.8423941552
Hartree =      24.8463331194
XC       =      -4.7944173767
NN       =           0.0000000000
-----
Total   =      57.5543736562
```

Result from Octopus:

```
***** SCF CYCLE ITER # 17 *****
etot = 5.75543034E+01 abs_ev = 4.97E-05 rel_ev = 6.14E-07
ediff = -5.09E-05 abs_dens = 6.73E-05 rel_dens = 8.42E-06
Matrix vector products: 31
Converged eigenvectors: 0

# State Eigenvalue [H] Occupation Error
1      9.056759      2.000000 (9.9E-06)
2     10.465603      2.000000 (1.2E-05)
3     10.465607      2.000000 (5.9E-06)
4     10.465608      2.000000 (7.1E-06)
```

Our result is similar to Octopus result.

### 6.2.5 Hydrogen atom

Result:

```

17      -0.4697222378    2.0241354490e-06    4.7656241007e-09
18      -0.4697214473    7.9055452079e-07    2.4859509631e-09

```

Convergence is achieved in 18 iterations

Eigenvalues:

```
1      -0.2449501107
```

-----  
Total energy components

-----  
Kinetic = 0.5027978737

Ps\_loc = -1.0263363191

Hartree = 0.2995317605

XC = -0.2457147624

NN = 0.0000000000

-----  
Total = -0.4697214473

Octopus result:

```
***** SCF CYCLE ITER # 12 *****
```

```
etot = -4.70593023E-01 abs_ev = 8.06E-07 rel_ev = 3.28E-06
```

```
ediff = -4.65E-11 abs_dens = 9.21E-06 rel_dens = 9.21E-06
```

```
Matrix vector products: 6
```

```
Converged eigenvectors: 0
```

```

# State Eigenvalue [H] Occupation Error
1      -0.245334    1.000000    (3.1E-06)

```

Quite similar to Octopus. Difference probably due to the Poisson solver.

Need to study convergence with respect to grid spacing.

### 6.2.6 H<sub>2</sub> molecule

Result:

```

15      -1.1975841508    4.0237235348e-06    3.8822972895e-08
16      -1.1975850469    8.9614081089e-07    2.1763673163e-08

```

Convergence is achieved in 16 iterations

Eigenvalues:

```
1      -0.3827444348
```

-----  
Total energy components

-----  
Kinetic = 1.1958829137

Ps\_loc = -3.7020236680

Hartree = 1.3006817627

XC = -0.6587927219

```
NN      =      0.6666666667
-----
Total   =     -1.1975850469
```

Octopus result:

```
***** SCF CYCLE ITER #   12 *****
etot = -1.19927504E+00 abs_ev  =  1.50E-06 rel_ev  =  1.95E-06
ediff =      1.68E-06 abs_dens =  2.11E-06 rel_dens =  1.06E-06
Matrix vector products:      15
Converged eigenvectors:      1

#  State  Eigenvalue [H]  Occupation   Error
   1      -0.383095      2.000000  (6.0E-07)
```

Similar observation as H atom.



## Chapter 7

# Kohn-Sham equation part II

Using nonlocal potential (pseudopotential)

In this work, we use norm-conserving pseudopotentials proposed by Goedecker-Teter-Hutter (GTH) [13]. These pseudopotentials can be written in terms of local  $V_{\text{loc}}^{\text{PS}}$  and angular momentum  $l$  dependent nonlocal components  $\Delta V_l^{\text{PS}}$ :

$$V_{\text{ene-nuc}}(\mathbf{r}) = \sum_I \left[ V_{\text{loc}}^{\text{PS}}(\mathbf{r} - \mathbf{R}_I) + \sum_{l=0}^{l_{\text{max}}} V_l^{\text{PS}}(\mathbf{r} - \mathbf{R}_I, \mathbf{r}' - \mathbf{R}_I) \right] \quad (7.1)$$

The local pseudopotential for  $l$ -th atom,  $V_{\text{loc}}^{\text{PS}}(\mathbf{r} - \mathbf{R}_I)$ , is radially symmetric function with the following radial form

$$V_{\text{loc}}^{\text{PS}}(r) = -\frac{Z_{\text{val}}}{r} \text{erf} \left[ \frac{\bar{r}}{\sqrt{2}} \right] + \exp \left[ -\frac{1}{2} \bar{r}^2 \right] (C_1 + C_2 \bar{r}^2 + C_3 \bar{r}^4 + C_4 \bar{r}^6) \quad (7.2)$$

with  $\bar{r} = r/r_{\text{loc}}$  and  $r_{\text{loc}}$ ,  $Z_{\text{val}}$ ,  $C_1$ ,  $C_2$ ,  $C_3$  and  $C_4$  are the corresponding pseudopotential parameters. In  $\mathbf{G}$ -space, the GTH local pseudopotential can be written as

$$V_{\text{loc}}^{\text{PS}}(G) = -\frac{4\pi}{\Omega} \frac{Z_{\text{val}}}{G^2} \exp \left[ -\frac{x^2}{2} \right] + \sqrt{8\pi^3} \frac{r_{\text{loc}}^3}{\Omega} \exp \left[ -\frac{x^2}{2} \right] \times \\ (C_1 + C_2(3 - x^2) + C_3(15 - 10x^2 + x^4) + C_4(105 - 105x^2 + 21x^4 - x^6)) \quad (7.3)$$

where  $x = G r_{\text{loc}}$ . The nonlocal component of GTH pseudopotential can be written in real space as

$$V_l^{\text{PS}}(\mathbf{r} - \mathbf{R}_I, \mathbf{r}' - \mathbf{R}_I) = \sum_{\mu=1}^{N_l} \sum_{\nu=1}^{N_l} \sum_{m=-l}^l \beta_{\mu lm}(\mathbf{r} - \mathbf{R}_I) h_{\mu\nu}^l \beta_{\nu lm}^*(\mathbf{r}' - \mathbf{R}_I) \quad (7.4)$$

where  $\beta_{\mu lm}(\mathbf{r})$  are atomic-centered projector functions

$$\beta_{\mu lm}(\mathbf{r}) = p_{\mu}^l(r) Y_{lm}(\hat{\mathbf{r}}) \quad (7.5)$$

and  $h_{\mu\nu}^l$  are the pseudopotential parameters and  $Y_{lm}$  are the spherical harmonics. Number of projectors per angular momentum  $N_l$  may take value up to 3 projectors. In  $\mathbf{G}$ -space, the nonlocal part of GTH pseudopotential can be described by the following equation.

$$V_l^{\text{PS}}(\mathbf{G}, \mathbf{G}') = (-1)^l \sum_{\mu}^3 \sum_{\nu}^3 \sum_{m=-l}^l \beta_{\mu lm}(\mathbf{G}) h_{\mu\nu}^l \beta_{\nu lm}^*(\mathbf{G}') \quad (7.6)$$

with the projector functions

$$\beta_{\mu lm}(\mathbf{G}) = p_{\mu}^l(G) Y_{lm}(\hat{\mathbf{G}}) \quad (7.7)$$

The radial part of projector functions take the following form

$$p_{\mu}^l(G) = q_{\mu}^l(Gr_l) \frac{\pi^{5/4} G^l \sqrt{r_l^{2l+3}}}{\sqrt{\Omega}} \exp\left[-\frac{1}{2} G^2 r_l^2\right] \quad (7.8)$$

For  $l = 0$ , we consider up to  $N_l = 3$  projectors:

$$q_1^0(x) = 4\sqrt{2} \quad (7.9)$$

$$q_2^0(x) = 8\sqrt{\frac{2}{15}}(3 - x^2) \quad (7.10)$$

$$q_3^0(x) = \frac{16}{3}\sqrt{\frac{2}{105}}(15 - 20x^2 + 4x^4) \quad (7.11)$$

For  $l = 1$ , we consider up to  $N_l = 3$  projectors:

$$q_1^1(x) = 8\sqrt{\frac{1}{3}} \quad (7.12)$$

$$q_2^1(x) = 16\sqrt{\frac{1}{105}}(5 - x^2) \quad (7.13)$$

$$q_3^1(x) = 8\sqrt{\frac{1}{1155}}(35 - 28x^2 + 4x^4) \quad (7.14)$$

For  $l = 2$ , we consider up to  $N_l = 2$  projectors:

$$q_1^2(x) = 8\sqrt{\frac{2}{15}} \quad (7.15)$$

$$q_2^2(x) = \frac{16}{3}\sqrt{\frac{2}{105}}(7 - x^2) \quad (7.16)$$

For  $l = 3$ , we only consider up to  $N_l = 1$  projector:

$$q_1^3(x) = 16\sqrt{\frac{1}{105}} \quad (7.17)$$

In the present implementation, we construct the local and nonlocal components of pseudopotential in the  $\mathbf{G}$ -space using their Fourier-transformed expressions and transformed them back to real space if needed. We refer the readers to the original reference [13] and the book [5] for more information about GTH pseudopotentials.

Due to the separation of local and non-local components of electrons-nuclei interaction, Equation (??) can be written as

$$E_{\text{ele-nuc}} = E_{\text{loc}}^{\text{PS}} + E_{\text{nloc}}^{\text{PS}} \quad (7.18)$$

The local pseudopotential contribution is

$$E_{\text{loc}}^{\text{PS}} = \int_{\Omega} \rho(\mathbf{r}) V_{\text{loc}}^{\text{PS}}(\mathbf{r}) d\mathbf{r} \quad (7.19)$$

and the non-local contribution is

$$E_{\text{nloc}}^{\text{PS}} = \sum_{\mathbf{k}} \sum_i w_{\mathbf{k}} f_{i\mathbf{k}} \int_{\Omega} \psi_{i\mathbf{k}}^*(\mathbf{r}) \left[ \sum_l \sum_{l=0}^{l_{\max}} V_l^{\text{PS}}(\mathbf{r} - \mathbf{R}_l, \mathbf{r}' - \mathbf{R}_l) \right] \psi_{i\mathbf{k}}(\mathbf{r}) d\mathbf{r}. \quad (7.20)$$

## Appendix A

# Introduction to Julia programming language

This chapter is intended to as an introduction to the Julia programming language.

This chapter assumes familiarity with command line interface.

### A.1 Installation

Go to <https://julialang.org/downloads/> and download the suitable file for your platform. For example, on 64 bit Linux OS, we can download the file `julia-1.x.x-linux-x86_64.tar.gz` where 1.x.x referring to the version of Julia. After you have downloaded the tarball you can unpack it.

```
tar xvf julia-1.x.x-linux-x86_64.tar.gz
```

After unpacking the tarball, there should be a new folder called `julia-1.x.x`. You might want to put this directory under your home directory (or another directory of your preference).

### A.2 Using Julia

#### A.2.1 Using Julia REPL

Let's assume that you have put the Julia distribution under your home directory. You can start the Julia interpreter by typing:

```
/home/username/julia-1.x.x/bin/julia
```

You should see something like this in your terminal:

```
$ julia

      _       _       _
     ( )     ( )     ( )
    _ _ _ _ | | _ _ _ _
    | | | | | | | / _ \ |
    | | | | | | | ( _ |
    _/ | \ _ ' | | | \ _ ' |
    | _/

Documentation: https://docs.julialang.org
Type "?" for help, "]?" for Pkg help.
Version 1.1.1 (2019-05-16)
Official https://julialang.org/ release
```

```
julia>
```

This is called the Julia REPL (read-eval-print loop) or the Julia command prompt. You can type the Julia program and see the output. This is useful for interactive exploration or debugging the program.

The Julia code can be typed after the `julia>` prompt. In this way, we can write Julia code interactively.

Example Julia session

```
julia> 1.2 + 3.4
4.6

julia> sin(2*pi)
-2.4492935982947064e-16

julia> sin(2*pi)^2 + cos(2*pi)^2
1.0
```

Using Unicode:

```
julia> α = 1234;

julia> β = 3456;

julia> α * β
4264704
```

The symbol  $\alpha$  can be entered in the Julia console by typing `\alpha` and then followed by typing Tab key. Another example is the  $\nabla$  symbol can be entered by typing `\nabla` followed by typing Tab key. Julia supports many Unicode symbols that can be used as names or operators. We can even use superscript symbol as in  $\nabla^2$ . A list of supported Unicode symbols in Julia can be found at <https://docs.julialang.org/en/v1/manual/unicode-input/>.

To exit the console we can type

```
julia> exit()
```

### A.2.2 Julia script file

We also can put the code in a text file with `.jl` extension and execute it with the command:

```
julia filename.jl
```

For example, type the following code in a file named `say_hello.jl`

```
function say_hello(name)
    println("Hello: ", name)
end
say_hello("efefer")
```

The run it by using the following command in the usual terminal session (not in a Julia console)

```
julia say_hello.jl
```

and the following output will be displayed in the terminal:



```
Hello: efefer
```

## A.3 Basic programming constructs

Julia has similarities with several popular programming languages such as Python, MATLAB, and R, to name a few.

### A.3.1 Displaying something

Using `println` and `@printf`

```
print("Hey")  
println("Hello")
```

```
using Printf  
@printf("Hey")  
@printf("Hello\n")
```

### A.3.2 Variables

A variable, in Julia, is a name associated (or bound) to a value. It's useful when you want to store a value (that you obtained after some math, for example) for later use. For example:

```
# Assign the value 10 to the variable x  
x = 10  
  
# Doing math with x's value  
x + 1  
  
# Reassign x's value  
x = 1 + 1  
  
# You can assign values of other types, like strings of text  
julia> x = "Hello World!"
```

Julia provides an extremely flexible system for naming variables. Variable names are case-sensitive, and have no semantic meaning (that is, the language will not treat variables differently based on their names).

Variable names must begin with a letter (A-Z or a-z), underscore, or a subset of Unicode code points greater than 00A0; in particular, Unicode character categories Lu/Ll/Lt/Lm/Lo/Nl (letters), Sc/So (currency and other symbols), and a few other letter-like characters (e.g. a subset of the Sm math symbols) are allowed. Subsequent characters may also include ! and digits (0-9 and other characters in categories Nd/No), as well as other Unicode code points: diacritics and other modifying marks (categories Mn/Mc/Me/Sk), some punctuation connectors (category Pc), primes, and a few other characters.

The only explicitly disallowed names for variables are the names of built-in statements:

While Julia imposes few restrictions on valid names, it has become useful to adopt the following conventions:

Names of variables are in lower case.

Word separation can be indicated by underscores ('\_'), but use of underscores is discouraged unless the name would be hard to read otherwise.

Names of Types and Modules begin with a capital letter and word separation is shown with upper camel case instead of underscores.

Names of functions and macros are in lower case, without underscores.

Functions that write to their arguments have names that end in !. These are sometimes called "mutating" or "in-place" functions because they are intended to produce changes in their arguments after the function is called, not just return a value.

### A.3.3 Mathematical operators

```
if a >= 1
    println("a is larger or equal to 1")
end
```

Example code 3

Plotting:

```
using PGFPlotsX
using LaTeXStrings
include("init_FD1d_grid.jl")
function my_gaussian(x::Float64; α=1.0)
    return exp( -α*x^2 )
end
function main()
    A = -5.0
    B = 5.0
    Npoints = 8
    x, h = init_FD1d_grid( A, B, Npoints )

    NptsPlot = 200
    x_dense = range(A, stop=5, length=NptsPlot)

    f = @pgf(
        Axis( {height = "6cm", width = "10cm" },
            PlotInc( {mark="none"}, Coordinates(x_dense, my_gaussian.(x_dense)) ),
            LegendEntry(L"f(x)"),
            PlotInc( Coordinates(x, my_gaussian.(x)) ),
            LegendEntry(L"Sampled $f(x)$"),
        )
    )
    pgfsave("TEMP_gaussian_1d.pdf", f)
end
main()
```

## Appendix B

### Lagrange basis functions

#### B.1 Lagrange-sinc function

$$L_i(x) = \frac{1}{\sqrt{h}} \frac{\sin[\pi(x - x_i)/h]}{\pi(x - x_i)/h} \quad (\text{B.1})$$

$$D_{jl}^{(2)} = \begin{cases} \frac{1}{h^2} \frac{\pi^2}{6} & \text{for } j = l \\ \frac{(-1)^{j-l}}{h^2} \frac{1}{(x_j - x_l)^2} & \text{for } j \neq l \end{cases} \quad (\text{B.2})$$

#### B.2 Periodic Lagrange function

For a given interval  $[0, L]$ , with  $L > 0$ , the grid points  $x_i$  appropriate for periodic Lagrange function are given by:

$$x_i = \frac{L}{2} \frac{2i - 1}{N} \quad (\text{B.3})$$

with  $i = 1, \dots, N$ . Number of points  $N$  should be an odd number.

The periodic cardinal functions  $L_i^{\text{per}}(x)$ , defined at grid point  $i$  are given by:

$$L_i^{\text{per}}(x) = \frac{1}{\sqrt{NL}} \sum_{n=1}^N \cos\left(\frac{\pi}{L}(2n - N - 1)(x - x_i)\right). \quad (\text{B.4})$$

The expansion of periodic function in terms of Lagrange functions:

$$f(x) = \sum_{i=1}^N c_i L_i^{\text{per}}(x) \quad (\text{B.5})$$

with expansion coefficients  $c_i = \sqrt{L/N} f(x_i)$ . When doing variational calculation, the coefficients  $c_i$  are the variational parameters. The actual function values  $f(x_i)$  at grid points  $x_i$  is obtained by  $f(x_i) = \sqrt{N/L} c_i$ . The prefactor is sometimes abbreviated by  $h = L/N$  and is also referred to as scaling factor.

Consider periodic potential in one dimension:

$$V(x + L) = V(x). \quad (\text{B.6})$$

Floquet-Bloch theorem states that the wave function solution for periodic potentials can be written in the form:

$$\psi_k(x) = e^{ikx} \phi_k(x) \quad (\text{B.7})$$

where function  $\phi_k(x)$  and its first derivative  $\phi'_k(x)$  have the same periodicity as  $V(x)$  and  $k$  is a constant called the crystal momentum. Substituting this expression to Schrodinger equation we obtain:

$$\left[ -\frac{\hbar^2}{2m} \left( \frac{d^2}{dx^2} + 2ik \frac{d}{dx} - k^2 \right) + V(x) \right] \phi_k(x) = E \phi_k(k). \quad (\text{B.8})$$

An alternative way of enforcing periodicity of the wave function is to require that:

$$\psi_k(x + L) = e^{ikL} \psi_k(x). \quad (\text{B.9})$$

This condition follows from:

$$\begin{aligned} \psi_k(x + L) &= e^{ik(x+L)} \phi_k(x + L) \\ &= e^{ik(x+L)} \phi_k(x) \\ &= e^{ikL} e^{ikx} \phi_k(x) \\ &= e^{ikL} \psi_k(x) \end{aligned}$$

Using periodic cardinal the Schrodinger equation for periodic potential can be written as:

$$\sum_{j=1}^N \left[ -\frac{\hbar^2}{2m} \left( D_{jl}^{(2)} + 2ik D_{jl}^{(1)} - k^2 \delta_{jl} \right) + V(j) \delta_{jl} \right] \phi(j) = E \phi(l) \quad (\text{B.10})$$

with  $l = 1, \dots, N$ .  $D_{jl}^{(1)}$  are matrix elements of the first derivatives:

$$D_{jl}^{(1)} = \begin{cases} 0 & j = l \\ -\frac{2\pi}{L} (-1)^{j-l} \left( 2 \sin \frac{\pi(j-l)}{N} \right)^{-1} & j \neq l \end{cases} \quad (\text{B.11})$$

and  $D_{jl}^{(2)}$  are matrix elements of the second derivatives,  $N' = (N - 1)/2$ :

$$D_{jl}^{(2)} = \begin{cases} -\left( \frac{2\pi}{L} \right)^2 \frac{N'(N' + 1)}{3} & j = l \\ -\left( \frac{2\pi}{L} \right)^2 (-1)^{j-l} \frac{\cos(\pi(j-l)/N)}{2 \sin^2[\pi(j-l)/N]} & j \neq l \end{cases} \quad (\text{B.12})$$

Note that,  $D_{jl}^{(1)}$  is not symmetric, but  $D_{jl}^{(1)} = -D_{lj}^{(1)}$ . Meanwhile, the second derivative matrix  $D_{jl}^{(2)}$  is symmetric, i.e.  $D_{jl}^{(2)} = D_{lj}^{(2)}$ . With the above expressions, first and second derivative of periodic cardinals can be expressed as

$$\frac{d}{dx} L_i^{\text{per}}(x) = \sum_{j=1}^N D_{ji}^{(1)} L_j^{\text{per}}(x) \quad (\text{B.13})$$

$$\frac{d^2}{dx^2} L_i^{\text{per}}(x) = \sum_{j=1}^N D_{ji}^{(2)} L_j^{\text{per}}(x) \quad (\text{B.14})$$

The previous approach also can be extended to periodic potential in 3D:

$$V(\mathbf{r}) = V(x, y, z) = V(x + L_x, y + L_y, z + L_z)$$

Using periodic LF, Schrodinger equation can be casted into the following form:

$$\left[ -\frac{\hbar^2}{2m} (\nabla^2 + 2i\mathbf{k} \cdot \nabla - \mathbf{k}^2) + V(\mathbf{r}) \right] \phi_k(\mathbf{r}) = E \phi_k(\mathbf{r}) \quad (\text{B.15})$$

### B.3 Cluster Lagrange function

For a given interval  $[A, B]$ , with  $B > A$ , the grid points  $x_i$  appropriate for cluster Lagrange function are given by:

$$x_i = A + \frac{B - A}{N + 1} i$$

where  $i = 1, \dots, N$ . Number of points  $N$  can be either odd or even number.

The cluster Lagrange functions  $L_i^{\text{clu}}(x)$ , defined at grid point  $i$  are given by:

$$L_i^{\text{clu}}(x) = \frac{2}{\sqrt{(N + 1)(B - A)}} \sum_{n=1}^N \sin(k_n(x_i - A)) \sin(k_n(x - A)). \quad (\text{B.16})$$

where  $k_n = \pi n / (B - A)$ . The expansion of a function  $f(x)$  in terms of cluster Lagrange functions:

$$f(x) = \sum_{i=1}^N c_i L_i^{\text{clu}}(x) \quad (\text{B.17})$$

with expansion coefficients  $c_i = \sqrt{(B - A)/(N + 1)} f(x_i)$ . When doing variational calculation, the coefficients  $c_i$  are the variational parameters. The actual function values  $f(x_i)$  at grid points  $x_i$  is obtained by  $f(x_i) = \sqrt{(N + 1)/(B - A)} c_i$ .

Matrix elements  $D_{jl}^{(2)}$  of the second derivatives for cluster Lagrange functions are

$$D_{jl}^{(2)} = \begin{cases} -\frac{1}{2} \left( \frac{\pi}{B - A} \right)^2 \frac{2(N + 1)^2 + 1}{3} - \frac{1}{\sin^2 [\pi j / (N + 1)]} & j = l \\ -\frac{1}{2} \left( \frac{\pi}{B - A} \right)^2 (-1)^{j-l} \left[ \frac{1}{\sin^2 \left[ \frac{\pi(j-l)}{2(N+1)} \right]} - \frac{1}{\sin^2 \left[ \frac{\pi(j+l)}{2(N+1)} \right]} \right] & j \neq l \end{cases} \quad (\text{B.18})$$

For free or cluster boundary condition, we don't need  $D_{jl}^{(1)}$ .



## Appendix C

# Introduction to Octopus DFT code

Prerequisites:

- Autotools and GNU Make
- C, C++ and Fortran compilers
- Libxc

```
autoreconf --install
./configure --prefix=path_to_install
make
make install
```

Example input:

```
CalculationMode = gs
Dimensions = 3
PeriodicDimensions = 0

TheoryLevel = dft
XCFunctional = lda_x + lda_c_vwn_1

BoxShape = parallelepiped
spacing = 0.4
%Lsize
 8 | 8 | 8
%

MixField = density
MixingScheme = Linear
Mixing = 0.5

FilterPotentials = filter_none
PseudopotentialSet = hgh_lda

% Coordinates
"H" | -0.75 | 0.0 | 0.0
"H" |  0.75 | 0.0 | 0.0
%
```

```
XCFunctional = none
```

```
TheoryLevel = independent_particles
```



# Bibliography

- [1] P. Hohenberg and W. Kohn. Inhomogenous electron gas. *Phys. Rev.*, 136:B864–871, 1964.
- [2] W Kohn and L.J. Sham. Self-consistent equations including exchange and correlation effects. *Phys. Rev.*, 140(4A):A1333–1138, 1965.
- [3] R.M. Martin. *Electronic Structure, Basic Theory and Practical Methods*. Cambridge University Press, Cambridge, UK, 2004.
- [4] Jorge Kohanoff. *Electronic structure calculations for solids and molecules: theory and computational methods*. Cambridge University Press, 2006.
- [5] Dominik Marx and Jürg Hutter. *Ab Initio Molecular Dynamics: Basic Theory and Advanced Methods*. Cambridge University Press, 2009.
- [6] Feliciano Giustino. *Materials Modelling using Density Functional Theory - Properties and Prediction*. Oxford University Press, 2014.
- [7] List of quantum chemistry and solid-state physics software. [https://en.wikipedia.org/wiki/List\\_of\\_quantum\\_chemistry\\_and\\_solid-state\\_physics\\_software](https://en.wikipedia.org/wiki/List_of_quantum_chemistry_and_solid-state_physics_software).
- [8] Julia programming language. <https://julialang.org/>.
- [9] Magnus R. Hestenes and Eduard Stiefel. Methods of conjugate gradients for solving linear systems. *Journal of Research of the National Bureau of Standards*, 49:409, 1952.
- [10] Jonathan Richard Shewchuk. An introduction to the conjugate gradient method without the agonizing pain. Available at <https://www.cs.cmu.edu/~quake-papers/painless-conjugate-gradient.pdf>.
- [11] Conjugate gradient method. [https://en.wikipedia.org/wiki/Conjugate\\_gradient\\_method](https://en.wikipedia.org/wiki/Conjugate_gradient_method).
- [12] Practical DFT mini-course. <http://jdftx.org/PracticalDFT.html>.
- [13] S. Goedecker, M. Teter, and J. Hutter. Separable dual-space Gaussian pseudopotentials. *Phys. Rev. B*, 54:1703–1710, 1996.